Includes easy-entry keystroke listings.

# HEWLETT-PACKARD

# HP-75

## USERS' LIBRARY SOLUTIONS

# Test Statistics

## NOTICE

The program material contained herein is supplied without representation or warranty of any kind. Hewlett-Packard Company therefore assumes no responsibility and shall have no liability, consequential or otherwise, of any kind arising from the use of this program material or any part thereof.

HEWLETT PACKARD

# HP-75 SOLUTIONS BOOK/MEDIA ORDER FORM

All HP-75 Solutions Books are available recorded from the Users' Library* or from any Hewlett-Packard Full Service Dealer. The individual Solutions Books are $12.50 each; recorded mini-data cassettes are $12.00 for each Solutions Book requested.

Telephone the Users' Library directly (503-757-2000, ext. 3371) with your Solutions Book request, your order will ship that SAME DAY; or, call our TOLL FREE number (800-547-3400, Continental U.S. only), and orders ship within 48 hours. Mail requests also ship within 48 hours of receipt.

**HP-75 Users' Library Solutions Books**

| | |
|---|---|
| 00075-90061 | Math I |
| 00075-90062 | Math II |
| 00075-90063 | Math III |
| 00075-90065 | Games |
| 00075-90066 | Electronics |
| 00075-90067 | Statistics |
| 00075-90070 | Real Estate |
| 00075-90071 | Finance |
| 00075-90076 | Test Statistics |

| Solutions Book Part Number | Title | Book Only | Mini-Data Cassettes | Price |
|---|---|---|---|---|
| 00075-90_____ | | | | |
| 00075-90_____ | | | | |
| 00075-90_____ | | | | |
| | **Please add Handling Charge on orders including Solutions Books** | | | 3.50 |
| | | | State and Local Taxes | |
| | | | **TOTAL** | |

All payments must be in U.S. dollars and drawn on a U.S. bank.
Send orders (with payment) directly to: HEWLETT-PACKARD USERS' LIBRARY
1000 N.E. Circle Blvd.
Corvallis, Oregon 97330

☐ Payment enclosed ☐ Master Card ☐ VISA ☐ Purchase order

Card Number/Company Purchase Order Exp

Name

Company (if applicable)

Address Telephone

City State/Country Zip

*Recorded Solutions Book programs shipped to U.S. customers only.

TABLE OF CONTENTS

# PROGRAM DESCRIPTION

## ONE SAMPLE TEST STATISTICS FOR THE MEAN

Suppose $(x_1,x_2,x_3,\ldots,x_n)$ is a sample from a normal population with a known variance $\sigma^2$ and an unknown mean $\mu$. A test of the null hypothesis

$$H_0:\mu = \mu_0$$

is based on the z statistic which has a standard normal distribution.

If the variance $\sigma^2$ is unknown then the t statistic, which has the t distribution with n-1 degrees of freedom, is used instead.

Equations:

$$z = \frac{\sqrt{n}\,(\bar{X}-\mu_0)}{\sigma}$$

$$t = \frac{\sqrt{n}\,(\bar{X}-\mu_0)}{S}$$

le mean and S is the population estimate of the
ation.

# SAMPLE PROBLEM

Suppose that the mean intelligence of the general population is 100 with a standard deviation of 15. Test the null hypothesis that the following sample of students is different from the population.

| Student | 1 | 2 | 3 | 4 | 5 |
|---|---|---|---|---|---|
| Score | 109 | 115 | 125 | 113 | 103 |

# SOLUTION

| STEP | INSTRUCTIONS | DISPLAY | INPUT |
|---|---|---|---|
| | Run "ONESAM" | One Sample Test For The Mean | |
| | | E to end input, C to correct | |
| 1 | Enter data as prompted | Enter datum 1? | 109 [RTN] |
| | | Enter datum 2? | 115 [RTN] |
| | | Enter datum 3? | 125 [RTN] |
| | **ERROR** | Enter datum 4? | 1 [RTN] |
| 2 | Call error correction | Enter datum 5? | C [RTN] |
| | | Datum 4 deleted = 1 | |
| 1 | Enter correct value | Enter datum 4? | 113 [RTN] |
| 1 | Continue | Enter datum 5? | 103 [RTN] |
| 3 | End data input | Enter datum 6? | E [RTN] |
| 4 | Enter general mean | Enter μo? | 100 [RTN] |
| 5 | Enter general std (N-1) | Enter σ? | 15 [RTN] |
| 6 | Read output. Use [RTN] to see | z = 1.9379 | [RTN] |
| | next output. Use [BACK] to | T = 3.5781 | [RTN] |
| | see last output | Mean = 113 | [RTN] |
| | | St. dev. = 8.124 | [RTN] |
| 7 | End program | Run, View, End, or Continue? | E [RTN] |

# USER INSTRUCTIONS

| STEP | INSTRUCTIONS | DISPLAY | INPUT |
|---|---|---|---|
| | Run "ONESAM" | One Sample Test For The Mean | |
| | | E to end input, C to correct | |
| | Do steps 1-2 N times | | |
| 1 | Enter value of indicated datum | Enter datum k? | Xk [RTN] |
| 2 | If an error was made: | Enter datum k? | C [RTN] |
| | This will be displayed: | Datum k deleted = Xk | |
| | If not finished with input, | | |
| | goto 1 | | |
| 3 | Indicate the end of input: | Enter datum k? | E [RTN[ |
| 4 | Enter μ naught: | $\mu_o$? | $\mu_o$ [RTN] |
| 5 | Enter σ: | σ? | σ [RTN] |
| 6 | Read output - [RTN] to see | z = | [RTN] |
| | next output, [BACK to see | T = | [RTN]/[BACK] |
| | last result | Mean = | [RTN]/[BACK] |
| | | St. dev = | [RTN]/[BACK] |
| 7 | Review routine | Run, View, End, or Continue? | |
| | R = rerun the program - step 1 | | R [RTN] |
| | V = review results - step 6 | | V [RTN] |
| | E = end the program | | E [RTN] |
| | C = goto step number 4 | | C [RTN] |

# VARIABLE NAMES

| NAME | DESCRIPTION | NAME | DESCRIPTION |
|---|---|---|---|
| A( ) | Data storage | K$ | General utility string |
| N | Sample size | S1 | Sum X = $\Sigma X$ |
| S2 | Sum X-squared $\Sigma(x^2)$ | M | Mean |
| S3 | Standard deviation (n-1) | Z | Z-value |
| T | t-value | U | Delay flag, = 1 if [BACK] |

# NOTES AND REFERENCES

Notes: 1. Both t and Z are computed. The user is reminded to use z only when the population standard deviation is known.

2. This program is limited to a maximum of 300 data points. To change this limit, change the dimension statement in line 80.

References: The formula for this program came from the HP-41C Users' Library Solutions book TEST STATISTICS.

# PROGRAM LISTING

```
 10 ! ONESAM - One sample
 20 ! Test statistics for
 30 ! the mean
 40 ! REV 11/01/82
 50 !
 60 DELAY .5
 70 DISP "One Sample Test for the Mean"
 80 DIM A(300),K$[100]
 90 DEF FNI(X) = INT(X*10000+.5)/10000          -Define precision of output
100 !
110 ! Delay routine
120 !
130 DEF FND                                     -Wait for 'RTN' or 'BACK' key.
                                                 Return 1 if 'BACK' key
140 K$=KEY$ @ IF K$<>CHR$(8) AND K$<>CH
    R$(13) THEN 140
150 FND=K$=CHR$(8) @ END DEF
160 !
170 ! Error correction
180 !
190 DEF FNE                                     -Delete incorrect input from
                                                 the counters
200 IF N<1 THEN DISP "Must have data to
     delete" @ BEEP 440 @ GOTO 230
210 DISP "Datum";N;"deleted =";A(N)
220 S2=S2-A(N)^2 @ S1=S1-A(N) @ N=N-1
230 FNE=0 @ END DEF
240 !
250 ! Initialize
260 !
270 N,S1,S2=0
280 DISP CHR$(197);" to end input, ";CH
    R$(195);" to correct" @ WAIT 1
290 !
300 ! Input loop
310 !
320 DISP "Enter datum";N+1; @ INPUT K$
330 IF UPRC$(K$)="E" AND N>1 THEN 410           -Goto end-of-data routine upon
                                                 entry of 'E'
340 IF UPRC$(K$)="E" THEN BEEP 220 @ DI
    SP "Must have more than 1 datum" @          -Display error if 'E' entered
    WAIT 1 @ GOTO 320                            and there is no data
350 IF UPRC$(K$)="C" THEN U=FNE @ GOTO
    320                                         -Call error correction routine
                                                 if 'C' is entered
360 ON ERROR BEEP 220 @ DISP 'Enter num         -Error trap
    eric, ["E"], or ["C"]' @ WAIT 1 @ G
    OTO 320
370 A(N+1)=VAL(K$) @ N=N+1 @ S1=S1+A(N)         -Increment counters and
     @ S2=S2+A(N)^2 @ GOTO 320                   continue adding data
380 !
390 ! Enter sigma, mu0
400 !
410 ON ERROR BEEP 220 @ DISP "Please en         -End-of-data routine, second
    ter numeric data" @ GOTO 420                 error trap
```

# PROGRAM LISTING

```
420 DISP "Enter ";CHR$(12);"o"; @ INPUT
     U0
430 DISP "Enter ";CHR$(9); @ INPUT S
440 !
450 ! Computation
460 !
470 OFF ERROR
480 M=S1/N @ S3=SQR((S2-S1^2/N)/N*(N/(N
    -1)))                              -Mean and standard
                                        deviation(n-1)
490 Z=SQR(N)*(M-U0)/S                  -Z-value and T-value
500 T=SQR(N)*(M-U0)/S3
510 PRINT "z=";FNI(Z) @ U=FND @ IF U TH -Display the results
    EN 510
520 PRINT "T=";FNI(T) @ U=FND @ IF U TH
    EN 510
530 PRINT "Mean =";FNI(M) @ U=FND @ IF
    U THEN 520
540 PRINT "St.dev. =";FNI(S3) @ U=FND @
     IF U THEN 530
550 !
560 ! Review routine
570 !
580 DISP CHR$(210);"un, ";CHR$(214);"ie -Review module
    w, ";CHR$(197);"nd, or ";CHR$(195);
    "ontinue ";
590 INPUT K$ @ K$=UPRC$(K$&" ")
600 ON POS("RVEC",K$[1,1])+1 GOTO 580,2
    70,510,610,410
610 STOP
```

# PROGRAM DESCRIPTION

## KENDALL'S COEFFICIENT OF CONCORDANCE

Kendall's Coefficient of Concordance is used to test the communality of preference among observers who have assigned n individuals ranks ranging from 1 to n according to some specified characteristic. The coefficient (W) varies from 0 (no community of preference) to 1 (perfect agreement), and is an extension of those rank-based measures used to test the degree of association in the two-variable case. This test is frequently considered a reliability measure of ranks.

$$W = \frac{12 \sum_{i=1}^{n} \left( \sum_{j=1}^{K} R_{ij} \right)^2}{(K^2)(n)(n^2-1)} - \frac{3(n+1)}{n-1}$$

$$\chi^2 = K(n-1)W$$

degrees of freedom = n-1

where: K is the number of observers

n is the number of individuals

$R_{ij}$ is the rank assigned to the ith individual by the jth observer

# SAMPLE PROBLEM

"Suppose three company executives are asked to inteview six job applicants and to rank them separately in their order of suitability for a job opening. The three independent sets of ranks given by executives X, Y, and Z to applicants a through f might be those shown . . . " (Siegel, NON-PARAMETRIC STATISTICS, p. 230).

Find the degree of agreement among the three executives whose rankings of six job applicants are shown below. (Artificial data).

| | Applicant | | | | | |
|---|---|---|---|---|---|---|
| | a | b | c | d | e | f |
| Executive X | 1 | 6 | 3 | 2 | 5 | 4 |
| Executive Y | 1 | 5 | 6 | 4 | 2 | 3 |
| Executive Z | 6 | 3 | 2 | 5 | 4 | 1 |
| $R_i$ (not computed) | 8 | 14 | 11 | 11 | 11 | 8 |

# SOLUTION

| STEP | INSTRUCTIONS | DISPLAY | INPUT |
|---|---|---|---|
| 1 | Run "KENDALL" | Kendall's Coeff. of Concordance | |
| 2 | Enter the # of executives | How many observers (K)? | 3 [RTN] |
| | Enter the # of applicants | How many subjects (N)? | 6 [RTN] |
| | | Type C to delete error | |
| 3 | Enter Appl.1, Exec.1 | Subject 1 Observer 1? | 1 [RTN] |
| | Enter Appl.1, Exec.2 | Subject 1 Observer 2? | 1 [RTN] |
| | Enter Appl.1, Exec.3 | Subject 1 Observer 3? | 6 [RTN] |
| | Enter Appl.2, Exec.1 | Subject 2 Observer 1? | 6 [RTN] |
| | | Subject 2 Observer 2? | 5 [RTN] |
| | | Subject 2 Observer 3? | 3 [RTN] |
| | Enter Applicant 3 | Subject 3 Observer 1? | 3 [RTN] |

# SOLUTION

| STEP | INSTRUCTIONS | DISPLAY | INPUT |
|---|---|---|---|
| | | Subject 3 Observer 2? | 6 [RTN] |
| | | Subject 3 Observer 3? | 2 [RTN] |
| | Enter Applicant 4 | Subject 4 Observer 1? | 2 [RTN] |
| | | Subject 4 Observer 2? | 4 [RTN] |
| | | Subject 4 Observer 3? | 5 [RTN] |
| | Enter Applicant 5 | Subject 5 Observer 1? | 5 [RTN] |
| | | Subject 5 Observer 2? | 2 [RTN] |
| | | Subject 5 Observer 3? | 4 [RTN] |
| | Enter Applicant 6 | Subject 6 Observer 1? | 4 [RTN] |
| | **Error** | Subject 6 Observer 2? | 334 [RTN] |
| | Call correction | Subject 6 Observer 3? | C [RTN] |
| | Correction displayed | 6,2 deleted = 334 | |
| | Enter correct value | Subject 6 Observer 2? | 3 [RTN] |
| | Continue | Subject 6 Observer 3? | 1 [RTN] |
| 4 | Coefficient of Concordance | W = .1619 | [RTN] |
| 5 | Chi-square value | Chi-square = 2.4286 | [RTN] |
| 6 | Degrees of freedon | Df = 5 | [RTN] |
| 7 | End program | Run again, View again, or End? | E [RTN] |

# USER INSTRUCTIONS

| STEP | INSTRUCTIONS | DISPLAY | INPUT |
|---|---|---|---|
| 1 | Run "KENDALL" | Kendall's Coeff. of Concordance | |
| 2 | Enter the # of observers | How many observers (K)? | k [RTN] |
| | Enter the # of subjects | How many subjects (n)? | n [RTN] |
| | | Type C to delete error | |
| 3 | Enter data as prompted | Subject i Observer j? | $R_{ij}$ [RTN] |
| | If an error was made: | Subject i Observer j? | C [RTN] |
| | Deleted datum displayed | i,j deleted = Rij | |
| | Goto 3 until done | | |
| 4 | Real output. Use [RTN] key | W = | [RTN] |
| | to see next, [BACK] to see | Chi-square = | [RTN]/[BACK] |
| | last result. | Df = | [RTN]/[BACK] |
| 5 | Review routine | Run again, View again, or End? | |
| | R = rerun program - step 2 | | R [RTN] or |
| | V = review results = step 4 | | V [RTN] or |
| | E = end program | | E [RTN] |

# VARIABLE NAMES

| NAME | DESCRIPTION | NAME | DESCRIPTION |
|---|---|---|---|
| K$ | General input string | A( ) | Temporary data storage |
| S0 | Grand sum of data | N | Grand N of data |
| S1 | $\Sigma Xij$ | S2 | $\Sigma(\Sigma Xij)^2$ |
| M | Mean of data | Z | Z statistic |
| E | ETA $\eta$ | | |

# NOTES AND REFERENCES

Notes: The value of the Coefficient of Concordance must be in the range of zero to 1 (0<=W<=1). If it is not within that range, the program will display a warning message. If you get the message "W is illegal. Check data", make sure that you have entered the data by subject (see sample problem), and that the observer's ranks are in the range 1 to n (the number of subjects). Tied ranks are each assigned the average of the ranks they would have been assigned had no ties occurred (see reference to Siegel).

References:

1. Siegel, Sidney, NONPARAMETRIC STATISTICS FOR THE BEHAVORIAL SCIENCES, (McGraw-Hill, 1956), p. 231-232.

2. Gibbond, J.D., NONPARAMETRIC STATISTICAL INFERENCE, (McGraw-Hill, 1971).

3. Conover, W.J., PRACTICAL NONPARAMETRIC STATISTICS (John Wiley, New York, 1971).

4. Formulae found in HP-41C Users' Library solutions TEST STATISTICS.

# PROGRAM LISTING

```
 10 ! KENDALL - Kendall's
 20 ! Coefficient of
 30 ! Concordance
 40 ! Rev 11/01/82
 50 !
 60 !
 70 !
 80 DELAY .5
 90 DISP "Kendall's Coeff. of Concordan
    ce"
100 DIM A(100),K$[100]
110 !
120 ! Delay routine
130 !
140 DEF FND                              -Wait for 'RTN' or 'BACK' keys.
                                          Return 1 if 'BACK' key
150 K$=KEY$ @ IF K$<>CHR$(8) AND K$<>CH
    R$(13) THEN 150
160 FND=K$=CHR$(8) @ END DEF
170 !
180 ! Error correction
190 !
200 DEF FNE                              -Error correction routine,
                                          decrement counters
210 IF J<2 THEN BEEP 220 @ DISP "Must h
    ave data to delete" @ GOTO 240
220 DISP STR$(I);",";STR$(J-1);" delete
    d =";A(J-1) @ WAIT 1
230 J=J-1 @ S0=S0-A(J)
240 FNE=0 @ END DEF
250 DEF FNI(X) = INT(X*10000+.5)/10000   -Define precision of output
260 !
270 ! Initialize
280 !
290 X0,X,S=0
300 INPUT "How many observers (K) ?";K   -Input dialogue and
                                          initialization
310 IF K<2 OR K>100 THEN BEEP 220 @ DIS
    P "Please enter 1<k<100" @ GOTO 300
320 INPUT "How many subjects (n) ?";N
330 IF N<2 THEN BEEP 220 @ DISP "Please
     enter n>1" @ GOTO 320
340 DISP "Type ";CHR$(195);" to delete
    error" @ WAIT 1
350 ON ERROR BEEP 220 @ DISP 'Enter num  -Error trap- display warning if
    eric data or ["C"]' @ WAIT 1 @ GOTO   invalid entry
     400
360 !
370 ! Data input
380 !
390 FOR I=1 TO N @ S0=0 @ FOR J=1 TO K
400 DISP "Subject";I;"Observer";J;
410 INPUT K$ @ IF UPRC$(K$)="C" THEN U=  -Call correction routine if
    FNE @ GOTO 400                        user enters 'C'
```

# PROGRAM LISTING

```
420 A(J)=VAL(K$) @ S0=S0+A(J)                    -Increment subject counter
430 NEXT J @ S=S+S0^2 @ NEXT I                   -End loop and increment total
                                                  counter
440 !
450 ! Calculate W,chi2,df
460 !
470 W=12*S/(K^2*N*(N^2-1))-3*(N+1)/(N-1          -Compute W
    )
480 C2=K*(N-1)*W @ D=N-1                         -Compute chi-square
490 !
500 ! Output routine
510 !
520 DISP "W=";FNI(W) @ U=FND @ IF U THE          -Printout routine
    N 520
530 IF W>1 OR W<0 THEN DISP W;"is illeg          -Display warning if W<0 or W>1
    al. Check data." @ BEEP 440,1.5 @ U
    =FND
540 DISP "Chi-square =";FNI(C2) @ U=FND
     @ IF U THEN 520
550 DISP "Df=";D @ U=FND @ IF U THEN 54
    0
560 DISP CHR$(210);"un again, ";CHR$(21          -Review routine
    4);"iew again, or ";CHR$(197);"nd";
570 INPUT K$ @ K$=UPRC$(K$&" ")
580 ON POS("RVE",K$[1,1])+1 GOTO 560,29
    0,520,590
590 STOP
```

# PROGRAM DESCRIPTION

## CORRELATION COEFFICIENT TEST

Under the assumptions of rank-order correlation analysis, the t statistic, which has the t distribution with n-2 degrees of freedom, may be used to test the null hypothesis (the true correlation coefficient $\rho=0$).

To test the null hypothesis $\rho = \rho_0$, where $\rho_0$ is a given number, the z statistic is used. z has approximately the normal distribution.

Equations:

$$t = \frac{r \quad n-2}{\sqrt{1-r^2}}$$

$$Z = \frac{\sqrt{n-3}}{2} \quad \ln \left[ \frac{(1+r)(1-\rho_0)}{(1-r)(1+\rho_0)} \right]$$

Where r is an estimate (based on a sample of size n) of the correlation coefficient $\rho$.

# SAMPLE PROBLEM

Given a sample size (N) of 31, and correlation coefficient of .12, test that the null hypothesis $\rho_0 = 0$.

# SOLUTION

| STEP | INSTRUCTIONS | DISPLAY | INPUT |
|---|---|---|---|
| | Run "CORRTEST" | Correlation Coefficient Test | |
| 1 | Enter the sample size | Enter the sample size (n)? | 31 [RTN] |
| 2 | Enter r | Correlation coefficient (r)? | .12 [RTN] |
| 3 | Enter $\rho_0$ | Enter RHO-naught? | 0 [RTN] |
| 4 | Read t | t = .650923 | [RTN] |
| 5 | Read Z | Z = .638055 | [RTN] |
| 6 | End program | Run again, View again, or End? | E [RTN] |

# USER INSTRUCTIONS

| STEP | INSTRUCTIONS | DISPLAY | INPUT |
|---|---|---|---|
| | Run "CORRTEST" | Correlation Coefficient Test | |
| 1 | Enter N | Enter the sample size (n)? | N [RTN] |
| 2 | Enter r | Correlation coefficient? | r [RTN] |
| 3 | Enter $\rho_0$ | Enter RHO-naught? | $\rho_0$ [RTN] |
| 4 | Read t value | t = t | [RTN] |
| 5 | Read Z value | Z = z | [RTN]/[BACK] |
| | Use [BACK] to review t-value | | |
| 6 | Review routine | Run again, View again, or End? | |
| | R = rerun the program - step 1 | | R [RTN] or |
| | V = review results - step 4 | | V [RTN] or |
| | E = end the program | | E [RTN] |

# VARIABLE NAMES

| NAME | DESCRIPTION | NAME | DESCRIPTION |
|---|---|---|---|
| K$ | Input string | N | Sample size |
| R | Correlation | RO | $P_o$ |
| T | t-score | Z | Z-score |
| U | Delay flag | | |

# NOTES AND REFERENCES

Notes:
1. This program will check that N>3, $\rho$ =<31, and $\rho_0$<1, and will display a warning if an error is found.
2. Usually, the z statistic is used when the sample size is large.
3. Note that both t and z are returned. If the test being made is that $\rho$ is not equal to zero then use z. Choice of the appropriate statistic is left to the user.

References:

1. Hogg and Craig, INTRODUCTION TO MATHEMATICAL STATISTICS, (Macmillan and Co., 1970).
2. J. Freund, MATHEMATICAL STATISTICS, (Prentice-Hall, 1971).
3. The formulae for this program come from the HP-41C Users' Library solutions TEST STATISTICS, "Test statistics for the correlation coefficient", p. 5.

# PROGRAM LISTING

```
 10 ! CORRTEST - Test
 20 ! statistics for the
 30 ! Correlation Coeff.
 40 ! REV 11/01/82
 50 !
 60 DELAY .5
 70 DISP "Correlation Coefficient Test"
 80 DIM K$[100]
 90 !
100 ! Delay function
110 !
120 DEF FND                                  -Wait for 'RTN' or 'BACK' keys-
                                              return 1 if 'BACK' key
130 K$=KEY$ @ IF K$()CHR$(8) AND K$()CH
    R$(13) THEN 130
140 FND=K$=CHR$(8) @ END DEF
150 DEF FNI(X) = INT(X*10^6+.5)/10^6         -Define output precision
160 !
170 ! Input
180 !
190 ON ERROR BEEP 220 @ DISP "Please en      -Error trap- display warning if
    ter numeric data" @ WAIT 1 @ GOTO 1       non-numeric entry
    90
200 INPUT "Enter the sample size (n) ?"      -Enter N
    ;N
210 IF N<3 THEN BEEP 220 @ DISP "Data e      -Verify that N>2
    rror - n<3" @ WAIT 1 @ GOTO 200
220 INPUT "Correlation coefficient (r)        -Enter correlation (r)
    ?";R
230 IF ABS(R)>=1 THEN BEEP 220 @ DISP "      -Verify that the absolute value
    Impossible correlation" @ GOTO 220        of (r) is less than one
240 INPUT "Enter RHO-naught ?";R0            -Enter rho-naught
250 IF ABS(R0)>=1 THEN DISP "Data error      -Verify that the absolute value
    . ";CHR$(12);"o illegal." @ GOTO 24       of rho is less than one
    0
260 !
270 ! Computation
280 !
290 T=R*SQR(N-2)/SQR(1-R^2)                 -Compute the t-value
300 Z=SQR(N-3)/2*LOG((1+R)*(1-R0)/((1-R      -Compute the z-value
    )*(1+R0)))
310 !
320 ! Print-out
330 !
340 DISP "t=";FNI(T) @ U=FND @ IF U THE      -Print the results
    N 340
350 DISP "Z=";FNI(Z) @ U=FND @ IF U THE
    N 340
360 !
370 ! Review routine
380 !
390 DISP CHR$(210);"un again, ";CHR$(21      -Review module
    4);"iew again, or ";CHR$(197);"nd";
400 INPUT K$ @ K$=UPRC$(K$&" ")
```

# PROGRAM LISTING

```
410 ON POS("RVE",K$[1,1])+1 GOTO 390,19
    0,340,420
420 END
```

# PROGRAM DESCRIPTION

## INTRACLASS CORRELATION COEFFICIENT

The intraclass correlation coefficient $r_I$ measures the degree of association among individuals within classes or groups. The coefficient is most easily calculated using the analysis of variance techniques. $r_I$ is the sample estimate of the population intraclass correlation coefficient $\rho_I$. If we can assume that the individuals within groups are random samples from normal populations with the same variance, then the hypothesis $\rho_I=0$ can be tested using the F statistic. This program also calculates the R-squared statistic, which is a measure of the relationship between the sample independent and dependent variables in the fixed effect case, and Omega-square, which is an estimate measure of the independent-dependent variable association in the population in the fixed effect case.

Mean of subjects in the $i^{th}$ sample

$$\bar{x}_i = \frac{\Sigma x_i}{n_i}$$

Standard deviation $\sigma_I = \sqrt{\frac{\Sigma(x_i^2)}{n_i} - (\bar{x}_i)^2}$

SS total $= \Sigma\left(\Sigma x_j^2\right) - \left(\Sigma\Sigma x\right)^2/n$

SS treatment $= \Sigma\left(\frac{T_i^2}{n_i}\right) - \frac{(\Sigma T)^2}{n}$ T is the column sum

SS error $= \Sigma\left(\Sigma x_j^2\right) - \Sigma\left(T_j^2/n_j\right)$

Df total = n-1 K = the number of sets
Df treatment = K-1 n = the total n
Df error = n-k J = the number of subjects

# PROGRAM DESCRIPTION

## INTRACLASS CORRELATION COEFFICIENT (continued)

MS treatment = SS treatment/df treatment

MS error = SS error/df error

F = MS treat/MS error

$R^2$ = SS treat/SS total

$\Omega^2$ = (SS treat - (MS error)(K-1))/(SS total + MS error)

$$\text{Intraclass } r = \left[\frac{\text{SS treat}}{K-1} - \frac{\text{SS error}}{K(J-1)}\right] \Big/ \left[\frac{\text{SS treat}}{K-1} + \frac{\text{SS error}}{K}\right]$$

# SAMPLE PROBLEM

Find the degree of association between individuals (A-B) within the following set of groups:

| Groups (sets) | Subjects A | Subjects B |
|---|---|---|
| 1 | 71 | 71 |
| 2 | 69 | 72 |
| 3 | 59 | 65 |
| 4 | 65 | 64 |
| 5 | 66 | 60 |
| 6 | 73 | 72 |
| 7 | 68 | 67 |
| 8 | 70 | 68 |

# SOLUTION

| STEP | INSTRUCTIONS | DISPLAY | INPUT |
|---|---|---|---|
| 1 | Run "INTRA" | Intraclass Correlation Coeff. | |
| 2 | Enter the N of groups | How many sets? | 8 [RTN] |
| 3 | Enter the N of subjects | How many subjects? | 2 [RTN] |
| | | C to correct error | |
| 4 | Enter set 1 | Set 1 Subject 1? | 71 [RTN] |
| | | Set 1 Subject 2? | 71 [RTN] |
| | Enter set 2 | Set 2 Subject 1? | 69 [RTN] |
| | | Set 2 Subject 2? | 72 [RTN] |
| | Enter set 3 | Set 3 Subject 1? | 59 [RTN] |
| | | Set 3 Subject 2? | 65 [RTN] |
| | Set 4: ERROR | Set 4 Subject 1? | 666 [RTN] |

# SOLUTION

| STEP | INSTRUCTIONS | DISPLAY | INPUT |
|---|---|---|---|
| | Call correction routine | Set 4 Subject 2? | C [RTN] |
| | Deleted value is displayed | (4,1) DELETED = 666 | |
| | Enter correct value | Set 4 Subject 1? | 65 [RTN] |
| | | Set 4 Subject 2? | 64 [RTN] |
| | Enter set 5 | Set 5 Subject 1? | 66 [RTN] |
| | | Set 5 Subject 2? | 60 [RTN] |
| | Enter set 6 | Set 6 Subject 1? | 73 [RTN] |
| | | Set 6 Subject 2? | 72 [RTN] |
| | Enter set 7 | Set 7 Subject 1? | 68 [RTN] |
| | | Set 7 Subject 2? | 67 [RTN] |
| | Enter set 8 | Set 8 Subject 1? | 70 [RTN] |
| | | Set 2 Subject 2? | 68 [RTN] |
| 4 | Read means | Set 1 Me = 71 Std = 0 | [RTN] |
| | | Set 2 Me = 70.5 Std = 1.5 | [RTN] |
| | | Set 3 Me = 62 Std = 3 | [RTN] |
| | | Set 4 Me = 64.5 Std - .5 | [RTN] |
| | | Set 5 Me = 63 Std = 3 | [RTN] |
| | | Set 6 Me = 72.5 Std = .5 | [RTN] |
| | | Set 7 Me = 67.5 Std = .5 | [RTN] |
| | | Set 8 Me = 69 Std = 1 | [RTN] |
| 5 | Read output | Total Me = 67.5 Std = 4.031 | [RTN] |
| | | Treat DF = 7 SS = 216 | [RTN] |
| | | Treat MS = 30.857 | [RTN] |
| | | Error DF = 8 SS = 260 | [RTN] |
| | | Error MS = 5.5 | [RTN] |

# SOLUTION

| STEP | INSTRUCTIONS | DISPLAY | INPUT |
|---|---|---|---|
| | | Total DF = 15 SS = 260 | [RTN] |
| | F value | F = 5.61039 | [RTN] |
| | | R squared = .830769 | [RTN] |
| | | Omega squared = .66855 | [RTN] |
| | Degree of association | Intraclass r = .697446 | [RTN] |
| 6 | End program | Run again, View again, or End? | E [RTN] |

# USER INSTRUCTIONS

| STEP | INSTRUCTIONS | DISPLAY | INPUT |
|---|---|---|---|
| 1 | Run "INTRA" | Intraclass Correlation Coefficient | |
| 2 | Enter the N of groups | How many sets? | k [RTN] |
| 3 | Enter the N of subjects | How many subjects? | n [RTN] |
| | | C to correct error | |
| 4 | Enter indicated datum | Set i Subject j? | Xij [RTN] |
| | If an error was made: | Set i Subject j? | C [RTN] |
| | This will be displayed: | (i,j) DELETED = Xij | |
| | Goto 4 until done | | |
| 5 | View means | Set i Me = Std = | [RTN] |
| 6 | View results. [RTN] to see | Total Me = Std = | [RTN] |
| | next result, [BACK] to see | Treat DF = SS = | [RTN]/[BACK] |
| | last result. | Treat MS = | [RTN]/[BACK] |
| | | Error DF = SS = | [RTN]/[BACK] |
| | | Error MS = | [RTN]/[BACK] |
| | | Total DF = SS = | [RTN]/[BACK] |
| | | F = | [RTN]/[BACK] |
| | | R squared = | [RTN]/[BACK] |
| | | Omega squared = | [RTN]/[BACK] |
| | | Intraclass r = | [RTN]/[BACK] |
| 7 | Review routine | Run again, View again, or End? | |
| | R = rerun program - step 2 | | R [RTN] |
| | V = review results - step 6 | | V [RTN] |
| | E = end program | | E [RTN] |

# VARIABLE NAMES

| NAME | DESCRIPTION | NAME | DESCRIPTION |
|---|---|---|---|
| T(i) | $\Sigma x_i$ | X2(i) | $\Sigma x_i^2$ |
| M(i) | Mean of treatment i | IO( ) | $t_i^2/n$ |
| S9(i) | SD of treatment i | V(i) | Variance of i |
| R( ) | Temporary storage | E1 | Value of deleted data |
| X$ | General input string | T | Grand sum of all data |
| N | Grand n of all data | X2 | $\Sigma x_i^2$ of treatment |
| IO | $\Sigma(T_i^2/n)$ | I1 | Intraclass correlation |
| S1 | Treatment SS | M1 | Treatment MS |
| S2 | Error SS | M2 | Error MS |
| S3 | Total SS | F | F-value |
| D1 | Treatment df | R2 | R squared |
| D2 | Error df | O2 | Omega squared |
| D3 | Total df | | |

# NOTES AND REFERENCES

Notes:

1. A maximum of 100 subjects per set may be used. To change this, change the dimension statements in line 70.

2. A maximum of 20 sets may be used. To change this, change lines 370 and 420.

3. Once a set has been finished, it may not be corrected.

References:

1. Winer, B.J., STATISTICAL PRINCIPLES IN EXPERIMENTAL DESIGN, (McGraw-Hill, New York, 1971) pages 210-214.

2. Hays, W.L., STATISTICS FOR PSYCHOLOGISTS, (Holt, Rinehart and Winston, 1963) page 382.

3. Ostle, B., STATISTICS, IN RESEARCH, Iowa State University Press, 1972.

4. HP-41C Users' Library, TEST STATISTICS, p. 33-34.

5. HP-75 STATISTICS SOLUTIONS BOOK, One-Way Analysis of Variance

# PROGRAM LISTING

```
 10 ! INTRAC-Intraclass
 20 ! Correlation Coeff.
 30 ! Rev 11/01/82
 40 !
 50 DELAY .5
 60 DISP "Intraclass Correlation Coeff.
    "
 70 DIM T(20),X2(20),I0(20),V(20),M(20)
    ,S9(20),R(100)
 80 DIM X$[20]
 90 !
100 ! Define variance
110 !
120 DEF FNS(S1,S2,N) = S2/N-(S1/N)^2      -Routine to calculate the
                                           variance
130 !
140 ! Delay routine
150 !
160 DEF FND                               -Wait for 'RTN' or 'BACK' key.
                                           Return 1 if 'BACK' key
170 A$=KEY$ @ IF A$="" THEN 170
180 FND=0 @ IF A$=CHR$(8) THEN FND=1
190 END DEF
200 !
210 ! Precision routine
220 !
230 DEF FNI(X) = INT(X*10^3+.5)/10^3      -Define precision of output
240 DEF FNJ(X) = INT(X*10^6+.5)/10^6
250 !
260 ! Correction
270 !
280 DEF FNE(C)                            -Error correction routine-
                                           decrement counters
290 IF 1=1 THEN BEEP @ DISP "Must have
    data to delete" @ WAIT 1 @ GOTO 330
300 E1=R(I-1) @ DISP "(";C;",";I-1;") D
    ELETED =";R(I-1) @ WAIT 1
310 T(C)=T(C)-E1 @ T=T-E1 @ I=I-1 @ N=N
    -1
320 X2=X2-E1^2 @ X2(C)=X2(C)-E1^2
330 FNE=0 @ END DEF
340 !
350 ! Initialize
360 !
370 FOR Y=1 TO 20 @ T(Y)=0 @ X2(Y)=0 @   -Set counters to zero
    I0(Y)=0 @ V(Y)=0 @ M(Y)=0 @ NEXT Y
380 N,T,X2,I0=0
390 !
400 ! Enter data
410 !
420 INPUT "How many sets ?"; K @ IF K<2
     OR K>20 THEN BEEP @ GOTO 420
430 INPUT "How many subjects? "; N0 @ I
    F N0<1 THEN BEEP @ GOTO 430
```

# PROGRAM LISTING

```
440 ON ERROR BEEP @ DISP "Please enter      -Error trap- return warning if
    numeric data" @ WAIT 1 @ GOTO 470        illegal data is entered
450 DISP CHR$(195);" to correct error."
     @ BEEP 440 @ WAIT 1
460 FOR J=1 TO K @ FOR I=1 TO N0            -Input loop
470 DISP 'Set';J;'Subject';I; @ INPUT X     -Call error correction
    $
480 IF X$="" THEN 470                       -Decrement counters
490 IF POS("Cc",X$) THEN U=FNE(J) @ GOT
    O 470
500 X=VAL(X$)
510 !
520 ! Loop counters
530 !
540 N=N+1 @ R(I)=X
550 T(J)=T(J)+X @ T=T+X @ X2(J)=X2(J)+X
    ^2 @ X2=X2+X^2
560 NEXT I
570 NEXT J
580 OFF ERROR
590 !
600 ! CALCULATION
610 ! Me,STD,Internal
620 !
630 FOR J=1 TO K
640 I0(J)=T(J)^2/N0 @ I0=I0+I0(J)           -Intermediate calculation
650 V(J)=FNS(T(J),X2(J),N0) @ S9(J)=SQR     -Variance and SD(n)
    (V(J))
660 M(J)=T(J)/N0                            -Mean
670 DISP "Set";J;"Me=";FNI(M(J));"Std="
    ;FNI(S9(J)) @ U=FND
680 NEXT J @ V=FNS(T,X2,N) @ M=T/N @ S9
    =SQR(V)
690 DISP "Total  Me=";FNI(M),"Std=";FNI
    (S9) @ U=FND
700 !
710 ! Int,SS,MS,DF
720 !
730 I1=T^2/N
740 S1=I0-I1 @ S2=X2-I0 @ S3=X2-I1          -Calculate sum-squares
750 D1=K-1 @ D2=N-K @ D3=N-1                -Calculate degrees of freedom
760 M1=S1/D1 @ M2=S2/D2                     -Calculate mean-squares
770 !
780 ! F,R^2,O^2
790 !
800 F=M1/M2 @ R2=S1/S3 @ O2=(S1-(K-1)*M
    2)/(S3+M2)
810 I1=(S1/(K-1)-S2/(K*(N0-1)))/(S1/(K-
    1)+S2/K)
820 !
830 ! Print Out
840 !
850 DISP "Treat  DF=";D1;"SS=";FNI(S1)
    @ U=FND
```

# PROGRAM LISTING

```
860 DISP "Treat  MS=";FNI(M1) @ U=FND @
     IF U THEN 850
870 DISP "Error  DF=";D2;"SS=";FNI(S2)
    @ U=FND @ IF U THEN 860
880 DISP "Error  MS=";FNI(M2) @ U=FND @
     IF U THEN 870
890 DISP "Total  DF=";D3;"SS=";FNI(S3)
    @ U=FND @ IF U THEN 880
900 DISP "F=";FNJ(F) @ U=FND @ IF U THE
    N 890
910 DISP "R squared=";FNJ(R2) @ U=FND @
     IF U THEN 900
920 DISP "Omega squared=";FNJ(O2) @ U=F
    ND @ IF U THEN 910
930 DISP "Intraclass r =";FNJ(I1) @ U=F
    ND @ IF U THEN 920
940 DISP CHR$(210);"un again, ";CHR$(21
    4);"iew again, or ";CHR$(197);"nd";
950 INPUT X$ @ ON POS("RVE",UPRC$(X$))+
    1 GOTO 940,370,690,960
960 STOP
```

-Review module

# PROGRAM DESCRIPTION

## KRUSKAL-WALLIS STATISTIC

Suppose we want to test the null hypothesis that k independent random samples of sizes $n_1, n_2, \ldots n_{k-1}, n_k$ come from identical continuous populations.

This program will arrange all values from k samples jointly (as if they were one sample) in an increasing order of magnitude. Values that have equal ranks will be assigned the average of the ranks that they cover. Let $R_{ij}$ be the rank of the jth value in the ith sample.

The Kruskal-Wallis statistic H can be used to test the null hypothesis that each observer has identical ranking preference.

When all sample sizes are large (>5), H is distributed approximately at the chi-square with k-1 degrees of freedom. For small samples, the test is based on special tables (not computed). This program will accept a maximum of 254 TOTAL observations, due to a restriction in the sorting routine.

Equation: df = K-1

$$H = \left(\frac{12}{n(n+1)}\right) \left( \sum_{i=1}^{K} \frac{\left(\sum_{j=1}^{n_i} R_{ij}\right)^2}{n_i} \right) - 3(N+1)$$

Where: $N = \sum_{i=1}^{K} n_i$

$R_{ij}$ = Rank of observation (i,j)

K = number of samples

$n_i$ = number of observations in the $i^{th}$ sample

# SAMPLE PROBLEM

Compute the H-statistic for the following data: (data from Conover, p. 258).

| 1 | | 2 | | 3 | | 4 | |
|---|---|---|---|---|---|---|---|
| Obs. | Rank | Obs. | Rank | Obs. | Rank | Obs. | Rank |
| 83 | 11 | 91 | 23 | 101 | 34 | 78 | 2 |
| 91 | 23 | 90 | 19.5 | 100 | 33 | 82 | 9 |
| 94 | 28.5 | 81 | 6.5 | 91 | 23 | 81 | 6.5 |
| 89 | 17 | 83 | 11 | 93 | 27 | 77 | 1 |
| 89 | 17 | 84 | 13.5 | 96 | 31.5 | 79 | 3 |
| 96 | 31.5 | 83 | 11 | 95 | 30 | 81 | 6.5 |
| 91 | 23 | 88 | 15 | 94 | 28.5 | 80 | 4 |
| 92 | 26 | 91 | 23 | | | 81 | 6.5 |
| 90 | 19.5 | 89 | 17 | | | | |
| | | 84 | 13.5 | | | | |

# SOLUTION

| STEP | INSTRUCTIONS | DISPLAY | INPUT |
|---|---|---|---|
| | Run "KRUSKA" | Kruskal-Wallis Statistic | |
| 1 | Enter the N of treatments | How many treatments? | 4 [RTN] |
| | | E to end treat., C to change | |
| 2 | Begin treatment 1 | Treat. 1 Subject 1? | 83 [RTN] |
| | | Treat. 1 Subject 2? | 91 [RTN] |
| | | Treat. 1 Subject 3? | 94 [RTN] |
| | | Treat. 1 Subject 4? | 89 [RTN] |
| | | Treat. 1 Subject 5? | 89 [RTN] |
| | | Treat. 1 Subject 6? | 96 [RTN] |
| | | Treat. 1 Subject 7? | 91 [RTN] |
| | | Treat. 1 Subject 8? | 92 [RTN] |

# SOLUTION

| STEP | INSTRUCTIONS | DISPLAY | INPUT |
|---|---|---|---|
| | | Treat. 1 Subject 9? | 90 [RTN] |
| 4 | End treatment 1 | Treat. 1 Subject 10? | E [RTN] |
| 2 | Begin treatment 2 | Treat. 2 Subject 1? | 91 [RTN] |
| | | Treat. 2 Subject 2? | 90 [RTN] |
| | | Treat. 2 Subject 3? | 81 [RTN] |
| | | Treat. 2 Subject 4? | 83 [RTN] |
| | | Treat. 2 Subject 5? | 84 [RTN] |
| | | Treat. 2 Subject 6? | 83 [RTN] |
| | | Treat. 2 Subject 7? | 88 [RTN] |
| | | Treat. 2 Subject 8? | 91 [RTN] |
| | | Treat. 2 Subject 9? | 89 [RTN] |
| | | Treat. 2 Subject 10? | 84 [RTN] |
| 4 | End treatment 2 | Treat. 2 Subject 11? | E [RTN] |
| 2 | Begin treatment 3 | Treat. 3 Subject 1? | 101 [RTN] |
| | | Treat. 3 Subject 2? | 100 [RTN] |
| | | Treat. 3 Subject 3? | 91 [RTN] |
| | **Error** | Treat. 3 Subject 4? | 9333 [RTN] |
| 3 | Call correction routine | Treat. 3 Subject 5? | C [RTN] |
| | This will be displayed: | (3,4) DELETED = 9333 | |
| 2 | Enter correct value | Treat. 3 Subject 4? | 93 [RTN] |
| | Continue | Treat. 3 Subject 5? | 96 [RTN] |
| | | Treat. 3 Subject 6? | 95 [RTN] |
| | | Treat. 3 Subject 7? | 94 [RTN] |
| 4 | End treatment 3 | Treat. 3 Subject 8? | E [RTN] |
| 2 | Begin treatment 4 | Treat. 4 Subject 1? | 93 [RTN] |

# SOLUTION

| STEP | INSTRUCTIONS | DISPLAY | INPUT |
|---|---|---|---|
| | | Treat. 4 Subject 2? | 82 [RTN] |
| | | Treat. 4 Subject 3? | 81 [RTN] |
| | | Treat. 4 Subject 4? | 77 [RTN] |
| | | Treat. 4 Subject 5? | 79 [RTN] |
| | | Treat. 4 Subject 6? | 81 [RTN] |
| | | Treat. 4 Subject 7? | 80 [RTN] |
| | | Treat. 4 Subject 8? | 81 [RTN] |
| 4 | End treatment 4 | Treat. 4 Subject 9? | E [RTN] |
| 5 | Skip the ranks | View the ranks (Y/N)? | N [RTN] |
| 6 | View printout | H = 25.46437 | [RTN] |
| | | df = 3 | [RTN] |
| | | Total n = 34 | [RTN] |
| 7 | End program | Run again, View again, or End? | E [RTN] |

# USER INSTRUCTIONS

| STEP | INSTRUCTIONS | DISPLAY | INPUT |
|---|---|---|---|
| | Run "KRUSKA" | Kruskal-Wallis Statistic | |
| 1 | Enter the number of treatments | How many treatments? | k [RTN] |
| | | E to end treat., C to correct | |
| | do steps 2-5 k times | | |
| 2 | Enter data as prompted | Treat. i Subject j? | Xij [RTN] |
| 3 | If an error was made: | Treat. i Subject j? | C [RTN] |
| | Deleted datum displayed: | (i,j) DELETED = Aij | |
| | Goto 2 until treatment done | | |
| 4 | Depress E to end treatment | Treat. i Subject j | E [RTN] |
| | Goto 2 until all treatments | | |
| | have been entered | | |
| 5 | Depress Y to see ranks | View the ranks (Y/N)? | Y or N [RTN] |
| | To view the ranks: | i,j S = R = | [RTN]/[BACK] |
| 6 | Read the printout. Use | H = | [RTN]/[BACK] |
| | [RTN] key to see next result, | df = | [RTN]/[BACK] |
| | [BACK] key to see last result. | Total n = | [RTN]/[BACK] |
| 7 | Review routine | Run again, View again, or End? | |
| | R = rerun program - step 1 | | R [RTN] |
| | V = review results - step 5 | | V [RTN] |
| | E = end program | | E [RTN] |

# VARIABLE NAMES

| NAME | DESCRIPTION | NAME | DESCRIPTION |
|---|---|---|---|
| A( ) | Data storage | N | Total N |
| X$ | General use string | A1( ) | Storage of ranks |
| N(j) | N of column j | N1 | Internal: total N |
| NO | Grand sum | K | Number of treatments |
| X | Temporary input value | H | H-value |
| R2 | $\Sigma(R,j)^2$ | R1 | Temporary sum of rank |
| D | Degrees of freedom | U* | Delay flag, = 1 if [BACK] |
| | | | |
| VARIABLES USED IN THE INDEX SORTING ROUTINE | | | |
| L | Lower parameter of search | U* | Upper parameter of search |
| I | Middle of binary search | K1 | Value to "search" for |
| T$ | Upper part of index string | B$ | Lower part of index string |
| A$ | Index string | | |
| | | | |
| VARIABLES USED IN THE RANKING AND TIED-VALUE CORRECTION ROUTINE | | | |
| L1 | Last discrete rank value | C1 | Index (position) of L1 |
| CO | Number of ranks that = L1 | C2 | Sum of indexes after C1 |
| T | "Current" rank value | R | Value to replace "current" tied ranks |
| | | | |
| * Some variables used for more than one purpose | | | |

# NOTES AND REFERENCES

Notes:

1. The program is presently limited to a maximum of 20 treatments. To change this limit, change lines 70 and 320.

2. The program is limited to a maximum of 254 total data items. To change this would require a major reprogramming effort, and would slow down the ranking routine.

References:

1. Conover, W.J., PRACTICAL NONPARAMETRIC STATISTICS (John Wiley and sons. 1971), p. 257-259.

2. Knuth, Donald E., THE ART OF COMPUTER PROGRAMMING, Volume 3, Sorting and Searching, (Addison-Wesley, 1972).

3. HP-41C Users' Library solutions, TEST STATISTICS, Kruskal-Wallis statistic.

# PROGRAM LISTING

```
 10 ! KRUSKAL: Kruskal-
 20 ! Wallis statistic
 30 ! REV 11/01/82
 40 !
 50 DELAY .5
 60 DISP "   Kruskal-Wallis Statistic"
 70 DIM A(255),A$[255],T$[255],B$[255],
    A1(255),X$[20],N(20)
 80 DEF FNC(X) = NUM(A$[X,X])                  -Conversion function for
                                                ranking routine
 90 !
100 ! Delay routine
110 !
120 DEF FND                                    -Wait for 'RTN' or 'BACK' keys-
                                                return 1 if 'BACK' key
130 A$=KEY$ @ IF A$="" THEN 130
140 FND=0 @ IF A$=CHR$(8) THEN FND=1
150 END DEF
160 !
170 ! Correction
180 !
190 DEF FNE(C)                                 -Correction routine
200 IF N(C)<1 THEN BEEP @ DISP "Must ha
    ve data to delete" @ GOTO 230
210 DISP "(";C;",";N(C);") DELETED =";A
    (N) @ WAIT 1
220 N(C)=N(C)-1 @ N=N-1 @ N0=N0-A(C)
230 FNE=0 @ END DEF
240 DEF FNI(X) = INT(X*10^5+.5)/10^5           -Define precision of output
250 !
260 ! Initialize
270 !
280 N1,R1,R2,N,N0=0
290 !
300 ! Enter data
310 !
320 INPUT "How many Treatments ?"; K @         -Enter and verify the number of
    IF K<2 OR K>20 THEN BEEP @ GOTO 320         treatments
330 DISP CHR$(197);" to end treat., ";C
    HR$(195);" to correct" @ WAIT 1
340 FOR J=1 TO K @ N(J)=0
350 DISP 'Treat.';J;'Subject';N(J)+1; @
     INPUT X$
360 IF UPRC$(X$)="E" THEN 490                  -Goto the end of the treatment
                                                if 'E' is entered
370 IF UPRC$(X$)="C" THEN U=FNE(J) @ GO        -Call error correction routine
    TO 350                                      if 'C' is entered
380 ON ERROR BEEP 220 @ DISP 'Enter num        -Error trap
    eric, ["E"], or ["C"]' @ WAIT 1 @ G
    OTO 350
390 X=VAL(X$)
400 !
410 ! Loop counters
420 !
```

# PROGRAM LISTING

```
430 N(J)=N(J)+1 @ N=N+1 @ A(N)=X @ N0=N   -Increment counters
    0+X
440 IF N)=254 THEN BEEP @ DISP "Can't E   -Check number of data items
    nter any more" @ WAIT 1 @ GOTO 490
450 GOTO 350
460 !
470 ! End of treatment
480 !
490 IF N(J)(2 THEN BEEP @ DISP "Must ha
    ve more than one subject" @ GOTO 35
    0
500 NEXT J
510 OFF ERROR
520 !
530 ! Rank data by
540 ! "Binary" sort
550 !
560 L1,A(N+1),A(0)=-INF @ C0=1 @ C1,C2=   -Binary index sort
    0
570 A$=CHR$(1)
580 FOR X=2 TO N                          -Binary search
590 L=1 @ U=X-1
600 I=INT((L+U)/2)
610 K1=A(NUM(A$[I,I]))
620 IF A(X)=K1 THEN 650
630 IF A(X)(K1 THEN U=I-1 ELSE L=I+1
640 IF U)=L THEN 600
650 IF 1)1 THEN B$=A$[1,I-1] ELSE B$=""
660 IF 1(X-1 THEN T$=A$[I+1,LEN(A$)] EL   -Insert present index into A$
    SE T$=""
670 IF A(X)(K1 THEN A$=B$&CHR$(X)&A$[I,
    LEN(A$)] ELSE A$=A$[1,I]&CHR$(X)&T$
680 NEXT X
690 A$=A$&CHR$(0)
700 !
710 ! Compute & store
720 ! ranks
730 !
740 FOR X=1 TO N                          -Store ranks in A()
750 T=A(FNC(X)) @ IF T()L1 AND C0=1 THE
    N L1=T @ A1(FNC(X))=X @ C1,C2=X @ G
    OTO 800
760 IF T=L1 THEN C0=C0+1 @ C2=C2+X        -Replace tied ranks
770 IF X()N AND T=L1 THEN 800
780 R=C2/C0 @ FOR Q=C1 TO C1+C0-1 @ A1(
    FNC(Q))=R @ NEXT Q
790 C0=1 @ L1=T @ A1(FNC(X))=X @ C1,C2=
    X
800 NEXT X
810 !
820 ! Compute r2/n
830 !
840 FOR X=1 TO K
850 R1=0 @ FOR Y=1 TO N(X)
```

# PROGRAM LISTING

```
 860 N1=N1+1 @ R1=R1+A1(N1) @ NEXT Y
 870 R2=R2+R1^2/N(X) @ NEXT X
 880 D=K-1                              -Compute H and degrees of
                                         freedom
 890 H=12/(N*(N+1))*R2-3*(N+1)
 900 !
 910 ! Print-out
 920 !
 930 DISP "View the ranks? (";CHR$(217);
     "/";CHR$(206);")";
 940 INPUT A$ @ A$=UPRC$(A$&" ")
 950 IF POS("NY",A$[1,1])=0 THEN 930
 960 IF A$="N " THEN 1050
 970 X1,X=0 @ Y=1
 980 PRINT "Sample";Y                   -Routine to allow user to view
                                         the ranks
 990 X=X+1 @ X1=X1+1 @ IF X>N THEN 1050
1000 IF X1>N(Y) THEN Y=Y+1 @ X1=0 @ X=X-
     1 @ GOTO 980
1010 PRINT STR$(Y);",";STR$(X1);" S=";A(
     X);"R=";A1(X) @ V=FND
1020 IF V AND X1>1 THEN X=X-1 @ X1=X1-1
     @ GOTO 1010
1030 IF V AND Y>1 THEN Y=Y-1 @ X1=N(Y) @
      X=X-1 @ GOTO 1010
1040 GOTO 990
1050 PRINT "H =";FNI(H) @ U=FND @ IF U T   -Printout
     HEN X1=N(K) @ X=N @ Y=K @ GOTO 1010
1060 PRINT "df=";D @ U=FND @ IF U THEN 1
     050
1070 PRINT "Total n=";N @ U=FND @ IF U T
     HEN 1060
1080 !
1090 ! Review routine
1100 !
1110 DISP CHR$(210);"un again, ";CHR$(21   -Review module
     4);"iew again, or ";CHR$(197);"nd "
     ;
1120 INPUT A$ @ A$=UPRC$(A$&" ")
1130 ON POS("RVE",A$[1,1])+1 GOTO 1110,2
     80,930,1140
1140 STOP
```

# PROGRAM DESCRIPTION

## MANN-WHITNEY U-TEST

This program calculates the Mann-Whitney test statistic on two independent samples of equal or unequal sizes. The Mann-Whitney test will test the null hypothesis that there is no difference between the two samples. The program will rank all values from both samples as if they were one sample, assigning tied ranks the mean of the positions that they occupy. This program will also accept, as input, data that has already been ranked in the preceding manner. The smaller of the two U-statistics will be displayed, as according to statistical convention. The Z-value that is displayed is approximately a random variable having the standard normal distribution.

For small samples (less than or equal to 8) the specially constructed tables should be used. For example: HANDBOOK OF STATISTICAL TABLES, D. B. Owen, Addison-Wesley, 1962.

Formula:

$$U = n_1 n_2 + \frac{n_1(n_1+1)}{2} - \sum_{i=1}^{n_1} R_i$$

$$z = \frac{U - \frac{n_1 n_2}{2}}{\sqrt{n_1 n_2 (n_1+n_2+1)/12}}$$

Where: $n_1$ = Size of sample 1

$n_2$ = Size of sample 2

$R_i$ = Rank assigned to item i of the sample with the smallest U.

Note: Both of the two possible U's are computed.
The smallest U is displayed, as is conventional.

# SAMPLE PROBLEM

The two-sample case is one in which the investigator has obtained two samples from possibly different populations. The null hypothesis tests the rejection of the assumption that the samples come from two different populations.

Consider two independent samples of students trained in a series of athletic events under two different conditions. The null hypothesis is that there are no differences between the training methods. Use the program MANN to calculate the ranks, U-statistic and z-statistic of these students.

| | Score | | | | |
|---|---|---|---|---|---|
| Method A | 18 | 15 | 13 | 21 | 11 |
| Method B | 13 | 6 | 2 | 5 | |

# SOLUTION

| STEP | INSTRUCTIONS | DISPLAY | INPUT |
|---|---|---|---|
| | Run "MANN" | Mann-Whitney U-Test Statistic | |
| | | Sample 1 C=change E=end | |
| 1 | Enter sample 1 | Sample 1 Item 1? | 18 [RTN] |
| | | Sample 1 Item 2? | 15 [RTN] |
| | | Sample 1 Item 3? | 13 [RTN] |
| | | Sample 1 Item 4? | 21 [RTN] |
| | | Sample 1 Item 5? | 11 [RTN] |
| 4 | End sample 1 | Sample 1 Item 6? | E [RTN] |
| | | Sample 2 C=change E=end | |
| 2 | Enter sample 2 | Sample 2 Item 1? | 13 [RTN] |
| | | Sample 2 Item 2? | 6 [RTN] |
| | **ERROR** | Sample 2 Item 3? | 299 [RTN] |
| 3 | Call error correction | Sample 2 Item 4? | C [RTN] |
| | This will be displayed | Item 3 deleted = 299 | |

# SOLUTION

| STEP | INSTRUCTIONS | DISPLAY | INPUT |
|---|---|---|---|
| 2 | Enter the correct value | Sample 2 Item 3? | 2 [RTN] |
| | Continue | Sample 2 Item 4? | 5 [RTN] |
| 4 | End sample 2 | Sample 2 Item 5? | E [RTN] |
| 5 | View the rank scores? | View the ranks? (Y/N)? | Y [RTN] |
| 6 | View the ranks. Use [RTN] | Sample 1 | |
| | to see the next rank. | 1,1 S = 18 R = 8 | [RTN] |
| | | 1,2 S = 15 R = 7 | [RTN] |
| | | 1,3 S = 13 R = 5.5 | [RTN] |
| | | 1,4 S = 21 R = 9 | [RTN] |
| | | 1,5 S = 11 R = 4 | [RTN] |
| | | Sample 2 | |
| | | 2,1 S = 13 R = 5.5 | [RTN] |
| | | 2,2 S = 6 R = 3 | [RTN] |
| | | 2,3 S = 2 R = 1 | [RTN] |
| | | 2,4 S = 5 R = 2 | [RTN] |
| 7 | View results, using [RTN] as | U = 1.5 | [RTN] |
| | above | Z = -2.08207 | [RTN] |
| 8 | End program | Run again, View again, or End? | E [RTN] |

# USER INSTRUCTIONS

| STEP | INSTRUCTIONS | DISPLAY | INPUT |
|---|---|---|---|
| | Run "MANN" | Mann-Whitney U-Test Statistic | |
| | | Sample 1 C = change, E = end | |
| 1 | Enter items of sample 1 | Sample 1 Item i? | A(1,i) [RTN] |
| | If an error was made: | Sample 1 Item i? | C [RTN] |
| | This will be displayed: | Item i deleted = A(1,i) | |
| | Goto 1 until done | | |
| | End sample 1 | Sample 1 Item i? | E [RTN] |
| | | Sample 2 C = change, E = end | |
| 2 | Enter items of sample 2 | Sample 2 Item i? | A(2,i) [RTN] |
| | If an error was made: | Sample 2 Item i? | C [RTN] |
| | Will not correct sample 1 | Item i deleted = A(2,i) | |
| | Goto 2 until done | | |
| | End sample 2 | Sample 2 Item i? | E [RTN] |
| 3 | Do you wish to view the | View the ranks? (Y/N)? | Y [RTN] or |
| | ranks? If not, goto 5 | | N [RTN] |
| 4 | View the ranks. Use the | Sample 1 | |
| | [RTN] key to see the next | 1,1 S = sample R = rank | [RTN] |
| | rank, [BACK] to see previous | 1,i S = sample R = rank | [RTN]/[BACK] |
| | rank | Sample 2 . . . . . | |
| 5 | View U and Z. Use [RTN] and | U = u value | [RTN]/[BACK] |
| | [BACK] as above | Z = z value | [RTN]/[BACK] |
| 6 | Review routine | Run again, View again, or End? | |
| | R = rerun the program - step 1 | | R [RTN] or |
| | V = review results - step 3 | | V [RTN] or |
| | E = end the program | | E [RTN] |

# VARIABLE NAMES

| NAME | DESCRIPTION | NAME | DESCRIPTION |
|---|---|---|---|
| A( ) | Data storage | A1( ) | Storage of ranks |
| N(j) | N of column j | N | Total N |
| X9 | Temporary input data value | Z | Z-statistic |
| U1,U2 | Large and small U statistic | K$ | General use string |
| R1,R2 | Sum of ranks in var. 1,2 | U* | Delay flag, = 1 if [BACK] |
| D | Degrees of freedom | | used |
| | | | |
| VARIABLES USED IN THE INDEX SORTING ROUTINE | | | |
| L | Lower parameter of search | U* | Upper parameter of search |
| I | Middle of binary search | K1 | Value to "search" for |
| T$ | Upper part of index string | B$ | Lower part of index |
| A$ | Index string | | |
| | | | |
| VARIABLES USED IN THE RANKING AND TIED-VALUE CORRECTION ROUTINE | | | |
| L1 | Last discrete rank value | C1 | Index (position) of L1 |
| C0 | Number of ranks that = L1 | C2 | Sum of indexes after C1 |
| T | "Current" rank value | R | Value to replace "current" tied ranks |
| | | | |
| * Some variables used for more than one purpose | | | |

# NOTES AND REFERENCES

Notes:

1. A maximum of 254 items (TOTAL) may be entered. This is due to the design of the ranking program, and can't be changed easily. See the Notes and References to Kruskal-Wallis statistic.

2. In accordance with statistical convention, the smaller of the two possible U-statistics is displayed. Variables U1 and U2 contain, respectively, the U-statistic computed using item 1 and the U-statistic computed using item 2.

References:

1. J.E. Freund, MATHEMATICAL STATISTICS, (Prentice-Hall, 1962).

2. S. Siegel, NON-PARAMETRIC STATISTICS, (McGraw-Hill, New York, 1956).

3. TEST STATISTICS, HP-41C Users' Library solutions, (MANN-WHITNEY STATISTIC).

# PROGRAM LISTING

```
 10 ! Mann-Whitney test statistic
 20 ! REV 11/01/82
 30 !
 40 DIM A(255),A$[255],T$[255],B$[255],
    A1(255),N(2)
 50 DELAY .5
 60 !
 70 ! Rank conversion
 80 !
 90 DEF FNC(X) = NUM(A$[X,X])                  -Conversion function for
                                                ranking routine
100 !
110 ! Error correction
120 !
130 DEF FNI(X) = INT(X*10^5+.5)/10^5           -Define precision of output
140 DEF FNE                                    -Delete user's input error and
                                                decrement counters
150 DISP "Item";N(S);"deleted =";A(N(S)
    +N0)
160 N=N-1 @ N(S)=N(S)-1 @ FNE=0
170 END DEF
180 !
190 ! Delay routine
200 !
210 DEF FND                                    -Wait for 'RTN' or 'BACK' keys.
                                                Return 1 if 'BACK' key
220 K$=KEY$ @ IF K$<>CHR$(13) AND K$<>C
    HR$(8) THEN 220
230 FND=K$=CHR$(8) @ END DEF
240 !
250 ! Initialize
260 !
270 DISP "Mann-Whitney U-test statistic
    "
280 ON ERROR BEEP 440 @ DISP 'Enter num   -Error trap
    eric, ["E"], or ["C"]' @ GOTO 350
290 L1,A(0)=-INF @ C0=1 @ N,N0,N(1),N(2
    ),R,C1,C2=0
300 !
310 ! Input loop
320 !
330 FOR S=1 TO 2
340 DISP "Sample ";S;"   ";CHR$(195);"=
    change, ";CHR$(197);"=end" @ WAIT 1
350 B$="" @ DISP "Sample";S;"Item";N(S)    -Enter sample value
    +1; @ INPUT B$
360 IF UPRC$(B$)="E" THEN 410              -End data entry if E is entered
370 IF UPRC$(B$)="C" AND N(S)>0 THEN U=    -Call correction routine if C
    FNE @ GOTO 350                          is entered
380 IF UPRC$(B$)="C" THEN DISP "Must ha    -Return error if attempting to
    ve data to delete" @ BEEP 220 @ GOT     delete non-existent data
    O 350
390 IF N>250 THEN BEEP @ DISP "WARNING-    -Return error if more than 254
     Can't accept n>254" @ WAIT 1.5         data items entered
```

# PROGRAM LISTING

```
400 X9=VAL(B$) @ N=N+1 @ N(S)=N(S)+1 @ A(N(S)+N0)=X9 @ GOTO 350          -Increment counters, store value, and continue
410 IF N(S)<2 THEN BEEP 220 @ DISP "Must have more than one item" @ GOTO 350
420 N0=N @ NEXT S
430 OFF ERROR
440 !
450 ! Binary sort
460 !                                                                    -Binary index sort
470 A$=CHR$(1) @ A(N+1)=-INF
480 FOR X=2 TO N                                                        -Binary search
490 L=1 @ U=X-1
500 I=INT((L+U)/2)
510 K1=A(NUM(A$[I,I]))
520 IF A(X)=K1 THEN 550
530 IF A(X)<K1 THEN U=I-1 ELSE L=I+1
540 IF U>=L THEN 500
550 IF I>1 THEN B$=A$[1,I-1] ELSE B$=""
560 IF 1<X-1 THEN T$=A$[I+1,LEN(A$)] ELSE T$=""                          -Insert preset index into A$
570 IF A(X)<K1 THEN A$=B$&CHR$(X)&A$[I,LEN(A$)] ELSE A$=A$[1,I]&CHR$(X)&T$
580 NEXT X
590 A$=A$&CHR$(0)
600 !
610 ! Paired ranks
620 !
630 FOR X=1 TO N                                                        -Store ranks in A1()
640 T=A(FNC(X)) @ IF T<>L1 AND C0=1 THEN L1=T @ A1(FNC(X))=X @ C1,C2=X @ GOTO 690
650 IF T=L1 THEN C0=C0+1 @ C2=C2+X                                       -Replace tied ranks
660 IF X<>N AND T=L1 THEN 690
670 R=C2/C0 @ FOR Q=C1 TO C1+C0-1 @ A1(FNC(Q))=R @ NEXT Q
680 C0=1 @ L1=T @ A1(FNC(X))=X @ C1,C2=X
690 NEXT X
700 !
710 ! Compute U and Z
720 !
730 R1,R2=0 @ FOR X=1 TO N(1) @ R1=R1+A1(X) @ NEXT X                    -Compute sum of ranks in variable 1 and 2
740 FOR X=N(1)+1 TO N(1)+N(2) @ R2=R2+A1(X) @ NEXT X
750 U2=N(1)*N(2)+N(2)*(N(2)+1)/2-R2                                      -Compute U using variable 2
760 U1=N(1)*N(2)+N(1)*(N(1)+1)/2-R1                                      -Compute U using variable 1
770 U=U1 @ IF U2<U1 THEN U=U2                                            -Choose smallest U
780 Z=(U-N(1)*N(2)/2)/SQR(N(1)*N(2)*(N(1)+N(2)+1)/12)                    -Compute z-score
790 !
800 ! View ranks
810 !
```

# PROGRAM LISTING

```
 820 DISP "View the ranks? (";CHR$(217);      -Routine to view the ranks
     "/";CHR$(206);")";
 830 INPUT K$ @ K$=UPRC$(K$&" ")
 840 IF POS("NY",K$[1,1])=0 THEN 820
 850 IF K$="N " THEN 970
 860 X1,X=0 @ Y=1
 870 DISP "Sample";Y
 880 X=X+1 @ X1=X1+1 @ IF X>N THEN 970
 890 IF X1>N(Y) THEN Y=Y+1 @ X1=0 @ X=X-
     1 @ GOTO 870
 900 DISP STR$(Y);",";STR$(X1);"  S=";A(
     X);"R=";A1(X) @ V=FND
 910 IF V AND X1>1 THEN X=X-1 @ X1=X1-1
     @ GOTO 900
 920 IF V AND Y>1 THEN Y=1 @ X1,X=N(1) @
      GOTO 900
 930 GOTO 880                                -Printout/display routine
 940 !
 950 ! View U,V
 960 !
 970 DISP "U = ";FNI(U) @ V=FND
 980 IF V THEN X=N @ X1=N(2) @ Y=2 @ GOT
     O 900
 990 DISP "Z=";FNI(Z) @ V=FND @ IF V THE
     N 970
1000 DISP CHR$(210);"un again, ";CHR$(21 -Review module
     4);"iew again, or ";CHR$(197);"nd";
1010 INPUT B$ @ B$=UPRC$(B$&" ")
1020 ON POS("RVE",B$[1,1])+1 GOTO 1000,2
     80,820,1030
1030 STOP
```

# PROGRAM DESCRIPTION

## FISHER'S EXACT PROBABILITY

Fisher's exact probability test is used for analyzing a 2 x 2 contingency table when the two independent samples are small in size.

| | |
|---|---|
| a (I-) | b (I+) |
| c (II-) | d (II+) |

Given frequencies a, b, c, d, with a as the smallest frequency, this program calculates:

1) the exact probability of observing the given frequencies in a 2 x 2 table, when the marginal totals are regarded as fixed, and

2) the exact probability $P_i$ (i=1, 2, ..., a) of each more extreme table having the same marginal totals.

Formula:

Exact probability $P_O = \frac{(a+b)!(c+d)!(a+c)!(b+d)!}{N!a!b!c!d!}$

Where: $N = a+b+c+d$

Each more extreme table (with the same margins)

$$P_i = \frac{(a+b)!(c+d)!(a+c)!(b+d)!}{N!(a-i)!(b+i)!(c+i)!(d-i)!}$$

Where: $P_i \in \{1,2,...a-1,a\}$

$X! = (1)(2)...(X-1)(X)$

# SAMPLE PROBLEM

Calculate $P_0$, $P_2$, $P_3$, and $P_4$ for the following table:

| Frequency: | | |
|---|---|---|
| | 7 | 10 |
| | 8 | 5 |

NOTE: The table must be arranged as:

| Frequency | | |
|---|---|---|
| | 5 | 8 |
| | 10 | 7 |

# SOLUTION

| STEP | INSTRUCTIONS | DISPLAY | INPUT |
|---|---|---|---|
| 1 | Run "FISHER" | Fisher Exact Probability Test | |
| 2 | Enter frequency A | Enter Group I- (A)? | 5 [RTN] |
| 3 | Enter frequency B | Enter Group I+ (B)? | 8 [RTN] |
| 4 | Enter frequency C | Enter Group II- (C)? | 10 [RTN] |
| 5 | Enter frequency D | Enter Group II+ (D)? | 7 [RTN] |
| 6 | Read probabilities, Use | Exact probability = .01467 | [RTN] |
| | [RTN] to see the next | Prob.( 1 ) = .05705 | [RTN] |
| | probability | Prob.( 2 ) = .13691 | [RTN] |
| | | Prob.( 3 ) = .1867 | [RTN] |
| | | Prob.( 4 ) = .12446 | [RTN] |
| | | Prob.( 5 ) = .02872 | [RTN] |
| 7 | End program | Run again, View again, or End? | E [RTN] |

# USER INSTRUCTIONS

| STEP | INSTRUCTIONS | DISPLAY | INPUT |
|---|---|---|---|
| | Arrange table so that a is | | |
| | the smallest frequency | | |
| 1 | Run "FISHER" | Fisher Exact Probability | |
| 2 | Enter frequency A | Enter group I- (A)? | a [RTN] |
| 3 | Enter frequency B | Enter group I+ (B)? | b [RTN] |
| 4 | Enter frequency C | Enter group II- (C)? | c [RTN] |
| 5 | Enter frequency D | Enter group II+ (D)? | d [RTN] |
| 6 | View probabilities. Use [RTN] | Exact probability = | [RTN] |
| | to see next display, [BACK] | Prob. (1) = | [RTN] |
| | to see previous one | Prob. (2) = | [RTN]/[BACK] |
| | | Prob. (3) = | [RTN]/[BACK] |
| | | Prob. (4) = | [RTN]/[BACK] |
| | | Prob. (5) = | [RTN]/[BACK] |
| 7 | Review routine | Run again, View again, or End? | |
| | R = rerun the program - step 2 | | R [RTN] |
| | V = review results - step 6 | | V [RTN] |
| | E = end program | | E [RTN] |

# VARIABLE NAMES

| NAME | DESCRIPTION | NAME | DESCRIPTION |
|---|---|---|---|
| P( ) | Probability storage | K$ | General input string |
| F | Factorial function | A | Group I- (A) |
| B | Group I+ (B) | C | Group II- (C) |
| D | Group II+ (D) | N | A+B+C+D |
| I,X | Index values | U | Delay flag, = 1 if [BACK] |

# NOTES AND REFERENCES

Notes: 1. a must be the smallest among the frequencies. Rearrange the table if necessary.

2. This program requires that a <= 50. To change this limit, change the dimension statement in line 70.

3. This program contains a routine for calculating the factorial of a number in lines 170-230. You can use it in your own program as follows:

```
1000 DEF FNF(N)
1010 F=1
1020 FOR X=2 to N
1030 F=F*X
1040 NEXT X
1050 FNF=F @ END DEF
 .......
2300 A=FNF(B)/C  (Sample formula)
 .......
```

References: 1. Sidney Siegel, NONPARAMETRIC STATISTICS, (McGraw-Hill, 1956).

2. Sir R.A. Fisher, STATISTICAL METHODS FOR RESEARCH WORKERS, (Oliver and Boyd, 1950).

3. Hewlett-Packard, HP-41C Users' Library Solutions TEST STATISTICS, formula from program FISHER'S EXACT TEST FOR A 2 x 2 CONTINGENCY TABLE, p. 39.

# PROGRAM LISTING

```
 10 ! FISHER - Fisher's exact
 20 ! test for a 2*2 contingency
 30 ! table
 40 ! REV 11/01/82
 50 !
 60 DISP "  Fisher Exact Probability Te
    st"
 70 DIM P(50),K$[20]
 80 !
 90 ! Delay routine
100 !
110 DEF FND                              -Wait for 'RTN' or 'BACK' keys.
                                          Return 1 if 'BACK' key
120 K$=KEY$ @ IF K$<>CHR$(13) AND K$<>C
    HR$(8) THEN 120
130 FND=K$=CHR$(8) @ END DEF
140 !
150 ! Calculate factorial
160 !
170 DEF FNF(N)                           -Function to compute factorial
180 F=1
190 FOR X=2 TO N
200 F=F*X
210 NEXT X
220 FNF=F
230 END DEF
240 !
250 ! Limit output precision
260 !
270 DEF FNI(X) = INT(X*100000+.5)/10000  -Function to define output
    0                                     precision
280 INPUT "Enter Group I- (A) ?";A       -Enter the frequencies
290 INPUT "Enter Group I+ (B) ?";B
300 INPUT "Enter Group II- (C) ?";C
310 INPUT "Enter group II+ (D) ?";D
320 IF (A<B)+(A<C)+(A<D)=3 THEN 340      -Verify that A is the smallest
                                          frequency
330 BEEP 880 @ DISP "A must be the smal
    lest freq." @ WAIT 1 @ GOTO 140
340 N=A+B+C+D
350 !
360 ! Calculate prob.
370 !
380 FOR I=0 TO A                         -Calculate the probabilities.
                                          Store exact prob. in A(0)
390 P(1)=FNF(A+B)*FNF(C+D)*FNF(A+C)*FNF
    (B+D)
400 P(I)=P(I)/(FNF(N)*FNF(A-I)*FNF(B+I)
    *FNF(C+1)*FNF(D-I))
410 NEXT I
420 !
430 ! Printout
440 !
450 DISP "Exact probability =";FNI(P(0)  -Display the probabilities
    ) @ U=FND @ IF U THEN 450
```

# PROGRAM LISTING

```
460 X=0
470 X=X+1 @ IF X>A THEN 540
480 DISP "Prob.(";X;") =";FNI(P(X)) @ U
    =FND
490 IF U THEN X=X-(X>1) @ GOTO 480
500 GOTO 470
510 !
520 ! Review routine
530 !
540 DISP CHR$(210);"un again, ";CHR$(21   -Review module
    4);"iew again, or ";CHR$(197);"nd "
    ;
550 INPUT K$ @ K$=UPRC$(K$&" ")
560 ON POS("RVE",K$[1,1])+1 GOTO 540,28
    0,450,570
570 STOP
```

# PROGRAM DESCRIPTION

## 2-FACTOR ANALYSIS OF VARIANCE

A two way analysis of variance allows the user to test the null hypothesis against columns, (condition a), rows (condition b), and the interaction of rows and columns, (a*b). This program can be generalized to any sized p*q 2-factor experiment, depending upon memory size. Unequal cell sizes are handled by considering each cell as though it contained the same number of subjects as all the other cells, with an adjustment based upon the harmonic mean. If the cell sizes are equal, the harmonic mean will have no effect.

RESTRICTIONS: If the cell sizes are relevant to the experimental manipulation, then other methods should be used. Severe variations in cell size (greater than 2:1) should be avoided. A negative Sum Square for the interaction term is cause for the immediate use of other methods.

p = number of rows

q = number of columns

| | Cell means | | | Total |
|---|---|---|---|---|
| | $\overline{X}_{11}$ | $\overline{X}_{12}$ .... | $\overline{X}_{1q}$ | $a_1$ |
| | $\overline{X}_{21}$ | $\overline{X}_{22}$ .... | $\overline{X}_{2q}$ | $a_2$ |
| | $\overline{X}_{31}$ | $\overline{X}_{32}$ .... | $\overline{X}_{3q}$ | $a_3$ |
| | ⋮ | ⋮ | ⋮ | ⋮ |
| | $\overline{X}_{p1}$ | $\overline{X}_{p2}$ .... | $\overline{X}_{pq}$ | $a_p$ |
| Total | $B_1$ | $B_2$ .... | $B_q$ | G |

Intermediate formulas:

$$(1) = G^2/pq$$

$$(2) = (\Sigma a_i^2)/q$$

$$(3) = (\Sigma B_j^2)/p$$

$$(4) = (\Sigma\Sigma \overline{X}_{ij})^2$$

$$\text{Harmonic mean } \overline{n}_h = \frac{pq}{\Sigma\Sigma(1/n_{ij})}$$

# PROGRAM DESCRIPTION

## 2-FACTOR ANALYSIS OF VARIANCE (continued)

$$SS_{within} = \Sigma\Sigma\left(\sum_{c} X^2_{cij} - \left(\frac{\Sigma X_{cij}}{n_{ij}}\right)^2\right) \qquad df_w = \Sigma\Sigma n_{ij} - pq \qquad MS_w = SS_w/df_w$$

$$SS_a = \bar{n}_h[(2)-(1)] \qquad df_a = p-1 \qquad MS_a = SS_a/df_a \qquad F_a = MS_a/MS_w$$

$$SS_b = \bar{n}_h[(3)-(1)] \qquad df_b = q-1 \qquad MS_b = SS_b/df_b \qquad F_b = MS_b/MS_w$$

$$SS_{ab} = \bar{n}_h[(4)-(2)-(3)+(1)] \qquad df_{ab} = (q-1)(p-1)$$

$$MS_{ab} = SS_{ab}/df_{ab} \qquad F_{ab} = MS_{ab}/MS_w$$

# SAMPLE PROBLEM

An experimenter wishes to study the effects of standard vs. reverse polish calculation under 3 conditions of instruction by scoring the number of errors per 100 calculations.

| | | Conditions (B) | | |
|---|---|---|---|---|
| | | 1 | 2 | 3 |
| Calculators (A) | RPN | 3,3,2 | 2,2,1 | 3,1,2 |
| | STANDARD | 9,6,7 | 8,6,6 | 9,5 |

(artificial data)

# SOLUTION

| STEP | INSTRUCTIONS | DISPLAY | INPUT |
|---|---|---|---|
| | Run "AOV2F" | 2-Factor Analysis of Variance | |
| 1 | Enter the row dimensions | How many rows (p)? | 2 [RTN] |
| 2 | Enter the column dimensions | How many columns (q)? | 3 [RTN] |
| 3 | Display the means | Print the means (Y/N)? | Y [RTN] |
| | | C to correct, E to end cell | |
| 6 | Enter column (1,1) | Row 1 Clm. 1 Item 1? | 3 [RTN] |
| | | Row 1 Clm. 1 Item 2? | 3 [RTN] |
| | | Row 1 Clm. 1 Item 3? | 2 [RTN] |
| 6b | End column (1,1) | Row 1 Clm. 1 Item 4? | E [RTN] |
| 7 | Read mean and SD (n-1) | [1,1] Mean = 2.66667 N = 3 | [RTN] |
| 7 | Use [BACK] to review mean | [1,1] Std. = .4714 | [RTN] |
| 6 | Enter column (1,2) | Row 1 Clm. 2 Item 1? | 2 [RTN] |
| | | Row 1 Clm. 2 Item 2? | 2 [RTN] |
| | | Row 1 Clm. 2 Item 3? | 1 [RTN] |
| 6b | End column (1,2) | Row 1 Clm. 2 Item 4? | E [RTN] |
| 7 | | [1,2] Mean = 1.66667 N = 3 | [RTN] |

# SOLUTION

| STEP | INSTRUCTIONS | DISPLAY | INPUT |
|---|---|---|---|
| | | [1,2] Std. = .4714 | [RTN] |
| 6 | Enter column (1,3) | Row 1 Clm. 3 Item 1? | 3 [RTN] |
| | | Row 1 Clm. 3 Item 2? | 1 [RTN] |
| | | Row 1 Clm. 3 Item 3? | 2 [RTN] |
| 6b | End column (1,3) | Row 1 Clm. 3 Item 4? | E [RTN] |
| 7 | | [1,3] Mean = 2 N = 3 | [RTN] |
| | | [1,3] Std. = .8165 | [RTN] |
| 6 | Enter column (2,1) | Row 2 Clm. 1 Item 1? | 9 [RTN] |
| | | Row 2 Clm. 1 Item 2? | 6 [RTN] |
| | | Row 2 Clm. 1 Item 3? | 7 [RTN] |
| 6b | End column (2,1) | Row 2 Clm. 1 Item 4? | E [RTN] |
| 7 | | [2,1] Mean = 7.3333 N = 3 | [RTN] |
| | | [2,1] Std. = 1.24722 | [RTN] |
| 6 | Enter column (2,2) | Row 2 Clm. 2 Item 1? | 8 [RTN] |
| | | Row 2 Clm. 2 Item 2? | 6 [RTN] |
| | | Row 2 Clm. 2 Item 3? | 6 [RTN] |
| 6b | End column (2,2) | Row 2 Clm. 2 Item 4? | E [RTN] |
| 7 | | [2,2] Mean = 6.66667 N = 3 | [RTN] |
| | | [2,2] Std. = .94281 | [RTN] |
| 6 | Enter column (2,3) | Row 2 Clm. 3 Item 1? | 9 [RTN] |
| | **Error** | Row 2 Clm. 3 Item 2? | 95 [RTN] |
| 6a | Call error correction | Row 2 Clm. 3 Item 3? | C [RTN] |
| | This will be displayed: | 2,3,2 deleted = 95 | |
| 6 | Enter correct value | Row 2 Clm. 3 Item 2? | 5 [RTN] |
| 6b | End column (2,3) | Row 2 Clm. 3 Item 3? | E [RTN] |

# SOLUTION

| STEP | INSTRUCTIONS | DISPLAY | INPUT |
|---|---|---|---|
| 7 | | [2,3] Mean = 7 N = 2 | [RTN] |
| | | [2,3] Std. = 2 | [RTN] |
| 9 | Read output using [RTN] key | A SS = 99.28205 df = 1 | [RTN] |
| | to see next result. | A MS = 99.28205 F = 58.50549 | [RTN] |
| | | B SS = 1.94872 df = 2 | [RTN] |
| | | B MS = .97436 F = .57418 | [RTN] |
| | | AB SS = .10256 df = 2 | [RTN] |
| | | AB MS = .05128 F = .03022 | [RTN] |
| | | Within SS = 18.66667 df = 11 | [RTN] |
| | | Within MS = 1.69697 | [RTN] |
| 10 | End program | Run again, View again, or End? | E [RTN] |

# USER INSTRUCTIONS

| STEP | INSTRUCTIONS | DISPLAY | INPUT |
|---|---|---|---|
| | Run "AOV2F" | 2-Factor Analysis of Variance | |
| 1 | Enter row dimensions | How many rows (p)? | p [RTN] |
| 2 | Enter column dimensions | How many columns (q)? | q [RTN] |
| 3 | Display cell means? | Print the means (Y/N)? | Y or N [RTN] |
| | | C to correct, E to end cell | |
| 4 | Enter cell value | Row i Clm. j Item c? | Xijc [RTN] |
| 5a | If error, call correction | Row i Clm. j Item c? | C [RTN] |
| | This will be displayed: | i,j,c deleted = Xijc | |
| | If not done, goto 4 else | | |
| 5b | If cell is finished | Row i Clm. j Item c? | E [RTN] |
| 6 | If you answered Y to #3: | [i,j] Mean = N= | [RTN] |
| | | [i,j] Std. = | [RTN]/[BACK] |
| 7 | Goto 6 until all rows and | | |
| | columns are done. | | |
| 8 | Read output. Use [RTN] key | A SS = df = | [RTN] |
| | to see next result, [BACK] | A MS = F = | [RTN]/[BACK] |
| | key to see previous result. | B SS = df = | [RTN]/[BACK] |
| | | B MS = F = | [RTN]/[BACK] |
| | | AB SS = df = | [RTN]/[BACK] |
| | | AB MS = F = | [RTN]/[BACK] |
| | | Within SS = df = | [RTN]/[BACK] |
| | | Within MS = | [RTN]/[BACK] |
| 9 | Review routine | Run again, View again, or End? | |
| | R = rerun the program - step 2<br>V = review results - step 8<br>E = End the program | | R [RTN]<br>V [RTN]<br>E [RTN |

# VARIABLE NAMES

| NAME | DESCRIPTION | NAME | DESCRIPTION |
|---|---|---|---|
| P | Number of rows (p) | Q | Number of columns (q) |
| A( ) | Sum of cell means (row) | X( ) | Temporary cell storage |
| B( ) | Sum of cell means (clm) | NO | Temporary n per cell |
| X$,A$ | General input strings | N1 | Grand total n |
| X1 | Temporary $\Sigma x$ per cell | X2 | Temporary $\Sigma x^2$ per cell |
| G1 | Grand sum of cell means | SO | Temporary SD(x) of cell |
| A2 | Sum of A-squared | B2 | Sum of B-squared |
| H1 | Harmonic mean | M | Temporary mean of cell |
| E1 | Computation formula #1 | E3 | Computation formula #3 |
| E4 | Computation formula #4 | E5 | Computation formula #5 |
| S1 | Sum-Squared of A (SSa) | M1 | Mean-Squared of A (MSa) |
| S2 | Sum-Squared of B (SSb) | M2 | Mean-Squared of B (MSb) |
| S3 | Sum-Squared of AB (SSab) | M3 | Mean-Squared of AB (MSab) |
| S4 | Sum-Squared within cell | M4 | Mean-Squared within cell |
| D1 | Df. of A (p-1) | D2 | Df. of B (q-1) |
| D3 | Df. of AB (q-1) | D4 | Df. within (G-P*Q) |
| F1 | F-ratio of A | F3 | F-ratio within |
| F2 | F-ratio of B | P1 | Flag 1=display cell means |

# NOTES AND REFERENCES

Notes: This program is limited to 20 rows, 20 columns, and 50 items per cell. To change this, change the dimensions in line 80.

References: Winer, B.J., STATISTICAL PRINCIPLES IN EXPERIMENTAL DESIGN, 2nd ED., (McGraw-Hill, New York, 1971), p. 446-447.

# PROGRAM LISTING

```
 10 ! ANOVA2F - 2-factor
 20 ! Analysis of variance
 30 ! (equal or unequal cell size)
 40 ! Unweighted (NON least-squares) me
    thod
 50 ! REV 11/01/82
 60 !
 70 DISP "2-Factor  Analysis of Varianc
    e" @ WAIT 1
 80 DIM A(20),B(20),X$[50],X(50),A$[20]
 90 !
100 ! Precision
110 !
120 DEF FNI(X) = INT(X*10^5+.5)/10^5      -Function to define output
                                           precision
130 !
140 ! Correction
150 !
160 DEF FNE                               -Error correction routine to
                                           delete data from counters
170 IF N0<1 THEN BEEP 880 @ DISP "Must
    Have data to delete" @ GOTO 200
180 DISP I;",";J;",";N0;"Deleted =";X(N
    0) @ WAIT 1
190 X1=X1-X(N0) @ X2=X2-X(N0)^2 @ N0=N0
    -1 @ N1=N1-1
200 FNE=0 @ END DEF
210 !
220 ! Delay
230 !
240 DEF FND                               -Wait for 'RTN' or 'BACK' keys.
                                           Returns 1 if 'BACK' key
250 X$=KEY$ @ IF X$<>CHR$(13) AND X$<>C
    HR$(8) THEN 250
260 FND=X$=CHR$(8) @ END DEF
270 !
280 ! Initialize
290 !
300 INPUT "How many rows (p) ?";P         -Starting prompts
310 INPUT "How many columns (q) ?";Q
320 INPUT "Print the means (Y/N) ?"; X$
     @ P1=POS(UPRC$(X$),"Y")
330 FOR X=1 TO P @ A(X)=0 @ NEXT X        -Initialize counters
340 FOR Y=1 TO Q @ B(Y)=0 @ NEXT Y
350 N1,G1,S4,E5,A2,B2,M1,N1,H1=0
360 DISP CHR$(195);" to correct, ";CHR$
    (197);" to end cell." @ WAIT 1
370 !
380 ! Error trapping
390 !
400 ON ERROR BEEP 880 @ DISP "Please en  -Error message for data entry
    ter [E],[C],or numeric" @ WAIT 1 @     routines
    GOTO 460
410 !
```

# PROGRAM LISTING

```
420 ! Input routine
430 !
440 FOR I=1 TO P @ FOR J=1 TO Q
450 N0=0 @ X1=0 @ X2=0
460 DISP "Row";I;"Clm.";J;"Item";N0+1;     -Data entry
470 X$="" @ INPUT X$
480 IF POS(UPRC$(X$),"C") THEN U=FNE @      -Call error correction if 'C'
    GOTO 460                                 is entered
490 IF POS(UPRC$(X$),"E") THEN 550          -Goto end of cell if 'E' is
                                             entered
500 X=VAL(X$) @ X1=X1+X @ X2=X2+X^2 @ N     -Increment cell counters and
    0=N0+1 @ X(N0)=X @ N1=N1+1               total N
510 GOTO 460
520 !
530 ! End cell
540 !
550 IF N0<2 THEN BEEP 880 @ DISP "Must
    have >1 datum per cell." @ WAIT 1 @
     GOTO 460
560 M=X1/N0 @ S4=S4+X2-X1^2/N0              -Cell mean
570 H1=H1+1/N0 @ A(I)=A(I)+M @ B(J)=B(J     -Harmonic mean counter, grand
    )+M                                      total
580 E5=E5+M^2 @ G1=G1+M @ S0=SQR((X2-X1     -Cell standard deviation (n)
    ^2/N0)/N0)
590 IF P1=0 THEN BEEP 440 @ GOTO 620        -Display mean if P1 is set
600 DISP "I";I;",";J;"] Mean =";FNI(M);
    "N=";N0 @ U=FND @ IF U THEN 600
610 DISP "I";1;",";J;"] Std.=";FNI(S0)
    @ U=FND @ IF U THEN 600
620 NEXT J @ NEXT I
630 H1=P*Q/H1
640 FOR I=1 TO P @ A2=A2+A(I)^2 @ NEXT      -Compute sum of A-squared, sum
    I                                        of B-squared
650 FOR J=1 TO Q @ B2=B2+B(J)^2 @ NEXT
    J
660 !
670 ! Sub-equations
680 !
690 E1=G1^2/(P*Q)
700 E3=A2/Q @ E4=B2/P
710 !
720 ! Compute SS,MS,df,F
730 !
740 D1=P-1 @ S1=H1*(E3-E1) @ M1=S1/D1       -Computation
750 D2=Q-1 @ S2=H1*(E4-E1) @ M2=S2/D2
760 D3=(Q-1)*(P-1) @ S3=H1*(E5-E3-E4+E1
    ) @ M3=S3/D3
770 D4=N1-P*Q @ M4=S4/D4
780 F1=M1/M4 @ F2=M2/M4 @ F3=M3/M4
790 !
800 ! Display the results
810 !
820 DISP "A  SS=";FNI(S1);"df=";D1 @ U=     -Display results
    FND @ IF U THEN 820
```

# PROGRAM LISTING

```
830 DISP "A  MS=";FNI(M1);"F=";FNI(F1)
    @ U=FND @ IF U THEN 820
840 DISP "B  SS=";FNI(S2);"  df=";D2 @
    U=FND @ IF U THEN 830
850 DISP "B  MS=";FNI(M2);"  F=";FNI(F2
    ) @ U=FND @ IF U THEN 840
860 DISP "AB SS=";FNI(S3);"  df=";D3 @
    U=FND @ IF U THEN 850
870 DISP "AB MS=";FNI(M3);"  F=";FNI(F3
    ) @ U=FND @ IF U THEN 860
880 DISP "Within SS=";FNI(S4);"df=";D4
    @ U=FND @ IF U THEN 870
890 DISP "Within MS=";FNI(M4) @ U=FND @
     IF U THEN 880
900 DISP CHR$(210);"un again, ";CHR$(21   -Review module
    4);"iew again, or ";CHR$(197);"nd";
910 INPUT A$ @ A$=UPRC$(A$&" ")
920 ON POS("RVE",A$[1,1])+1 GOTO 900,30
    0,820,930
930 STOP
```

# PROGRAM DESCRIPTION

## BARTLETT'S CHI-SQUARE STATISTIC

Bartlett's Chi-square has a distribution (approximately) with k-1 degrees of freedom which can be used to test the null hypothesis that the variances are all estimates of the same population variance.

F max tests the hypothesis that the largest and the smallest variance come from the same population of variances.

Formula: (Note: $\chi$ = CHI)

$$\chi^2 = \frac{f \ln (S^2) - \sum_{i=1}^{K} f_i \ln(s_i^2)}{1+\left(\frac{1}{3(K-1)}\right)\left[\left(\sum\frac{1}{f_i}\right) - \frac{1}{f}\right]}$$

Where: $s_i^2$ = Sample variance of the $i^{th}$ sample

$F_i$ = Degrees of freedom $(n_i-1)$ of the $i^{th}$ sample

$$s^2 = \frac{\sum_{i=1}^{K} f_i S_i^2}{f}$$

$$f = \sum_{i=1}^{K} f_i$$

$$F \text{ max} = \max (s^2)/\min(s^2)$$

# SAMPLE PROBLEM

Determine whether the variances of the groups shown below differ significantly across the groups. The obtained chi-square of 1.04 shows that the differences are not significant.

| Group | N (not used) | Variance ($\sigma$) | Degrees of freedom (df) |
|---|---|---|---|
| 1 | 21 | 12.20 | 20 |
| 2 | 13 | 13.50 | 12 |
| 3 | 15 | 7.86 | 14 |
| 4 | 10 | 10.89 | 9 |

Df is based upon N-1

Data comes from Edwards, EXPERIMENTAL DESIGN IN PSYCHOLOGICAL RESEARCH.

# SOLUTION

| STEP | INSTRUCTIONS | DISPLAY | INPUT |
|---|---|---|---|
| | Run "BARTLETT" | Bartlett's Chi-square Statistic | |
| | | E,E to end, C,C to correct | |
| 1 | Enter sample 1 | Sample 1 enter var., df? | 12.2, 20 [RTN] |
| | Enter sample 2 | Sample 2 enter var., df? | 13.5, 12 [RTN] |
| | **Error** | Sample 3 enter var., df? | 12,3 [RTN] |
| 2 | Call error correction | Sample 4 enter var., df? | C C [RTN] |
| | This will be displayed: | Sample 3 deleted = 12,3 | |
| 1 | Enter correct values | Sample 3 enter var., df? | 7.86, 14 [RTN] |
| | Enter sample 4 | Sample 4 enter var., df? | 10.89, 9 [RTN] |
| 3 | End data input | Sample 5 enter var., df? | E,E [RTN] |
| 4 | Real results. Use [RTN] to | Chi-square = 1.04955 | [RTN] |
| | see next item. | Fmax = 1.71756 | [RTN] |
| | | df = 3 | [RTN] |
| 5 | End program | Run again, View again, or End? | E [RTN] |

# USER INSTRUCTIONS

| STEP | INSTRUCTIONS | DISPLAY | INPUT |
|---|---|---|---|
| 1 | Run "BARTLETT" | Bartlett's Chi-square Statistic | |
| | | E,E to end, C,C to correct | |
| 2 | Enter value of sample | Sample i enter var, df? | σ,df. [RTN] |
| 3 | If an error was made: | Sample i enter var, df? | C,C [RTN] |
| | this will be displayed | Sample i deleted = Si,DFi | |
| | Goto 2 until all samples have | | |
| | been entered | | |
| 4 | To end data input: | Sample i enter var, df? | E,E [RTN] |
| 5 | Read display. Use [RTN] to | Chi-square = | [RTN] |
| | see next item, [BACK] to | F max = | [RTN]/[BACK] |
| | see previous item | df = | [RTN]/[BACK] |
| 6 | Review routine | Run again, View again, or End? | |
| | R = rerun the program - step 2 | | R [RTN] |
| | V = review results - step 5 | | V [RTN] |
| | E = end the program | | E [RTN] |

# VARIABLE NAMES

| NAME | DESCRIPTION | NAME | DESCRIPTION |
|---|---|---|---|
| S2( ) | Variance storage | F( ) | Storage of df. |
| A$ | Variance input, general use | B$ | Df. input |
| K | Total number of sample | U | Delay flag, = 1 if [BACK] used |
| S4 | Sum of inverse df | M1 | Minimum variance |
| A1 | Maximum variance | MO | F max |
| CO | Chi-square value | D | Total df. (degrees of freedom) |

# NOTES AND REFERENCES

Notes:

1. This program is presently limited to a maximum of 50 samples. To change this limit, change the dimension statement in line 70.

2. Because a different method was used, the solution to the sample problem is slightly different from the solution in Edwards EXPERIMENTAL DESIGN IN PSYCHOLOGICAL RESEARCH, p. 198.

References:

1. Hald, A., STATISTICAL THEORY WITH ENGINEERING APPLICATIONS, (John Wiley and Sons, 1960).

2. Edwards, A., EXPERIMENTAL DESIGN IN PSYCHOLOGICAL RESEARCH, (Rinehart & Co., 1950), p. 198.

3. Hewlett-Packard, HP-41C Users' Library solutions TEST STATISTICS, Procedure from Bartlett's Chi-square Statistic, p. 46.

# PROGRAM LISTING

```
 10 ! BARTLETT - Bartlett's Chi-
 20 ! square statistic (test for
 30 ! homogeneity of variance)
 40 !
 50 ! REV 11/01/82
 60 DISP "Bartlett's Chi-square Statist
    ic" @ WAIT 1
 70 DIM S2(50),F(50),A$[10],B$[10]
 80 !
 90 ! Correction routine
100 !
110 DEF FNE                              -Correct user's output
120 IF K<1 THEN BEEP 440 @ DISP "Must h
    ave data to delete" @ GOTO 150
130 DISP "Sample";K;"deleted =";S2(K);"
    ,";F(K)
140 S2=S2-S2(K)*F(K) @ F=F-F(K) @ S3=S3  -Delete incorrect value from
    -F(K)*LOG10(S2(K)) @ S4=S4-1/F(K) @    the counters
     K=K-1
150 FNE=0 @ END DEF
160 !
170 ! Delay routine
180 !
190 DEF FND                              -Wait for 'RTN' or 'BACK' keys.
                                          Return 1 if 'BACK' key
200 K$=KEY$ @ IF K$<>CHR$(8) AND K$<>CH
    R$(13) THEN 200
210 FND=K$=CHR$(8) @ END DEF
220 DEF FNI(X) = INT(X*10^5+.5)/10^5     -Function to define the output
                                          precision
230 S2,S3,S4,F,K=0 @ A1=-INF @ M1=INF    -Initialize counters
240 DISP "E,E to end, C,C to correct" @
     WAIT 1
250 ON ERROR BEEP 220 @ DISP "Enter [E]  -Error trap
    [RTN],[C][RTN] or S2,F" @ WAIT 1 @
    GOTO 260
260 DISP "Sample";K+1;"enter var., df";
270 INPUT A$,B$
280 IF POS(UPRC$(A$[1,1]),"E") THEN 360  -Goto end-of-data if 'E' is
                                          entered
290 IF POS(UPRC$(A$[1,1]),"C") THEN U=F  -Call error correction if 'C'
    NE @ GOTO 260                         is entered
300 S2(K+1)=VAL(A$) @ F(K+1)=VAL(B$)     -Increment counters and enter
                                          next sample
310 S3=S3+VAL(B$)*LOG10(VAL(A$)) @ S4=S
    4+1/VAL(B$)
320 F=F+VAL(B$) @ S2=S2+VAL(A$)*VAL(B$)
     @ K=K+1 @ GOTO 250
330 !
340 ! End of data
350 !
360 S2=S2/F @ C0=(F*LOG10(S2)-S3)*2.302  -End of data- compute
    6/(1+1/(3*(K-1))*(S4-1/F))            chi-square value
370 FOR I=1 TO K @ IF S2(I)>A1 THEN A1=   -Determine maximum/minimum
    S2(I)                                 variance
```

# PROGRAM LISTING

```
380 IF S2(I)<M1 THEN M1=S2(I)
390 NEXT I
400 M0=A1/M1 @ D=K-1

410 PRINT "Chi-square =";FNI(C0) @ U=FN
    D @ IF U THEN 410
420 PRINT "Fmax =";FNI(M0) @ U=FND @ IF
     U THEN 410
430 PRINT "df=";D @ U=FND @ IF U THEN 4
    20
440 DISP CHR$(210);"un again, ";CHR$(21
    4);"iew again, or ";CHR$(197);"nd "
    ;
450 INPUT "?"; A$ @ A$=UPRC$(A$&" ")
460 ON POS("RVE",A$[1,1])+1 GOTO 440,23
    0,410,470
470 STOP
```

-Compute F max, total degrees of freedom
-Display the results

-Review module

# PROGRAM DESCRIPTION

## DIFFERENCES AMONG PROPORTIONS

This program tests proportions in independent sets of data to determine if each could have been randomly drawn from the same population of proportions. A chi-square statistic with k-1 degrees of freedom is computed. Theta is a measure of association between the independent (groups) and dependent (proportions) variables.

Equations:

$$\chi^2 = \sum_{i=1}^{K} \frac{(C_{1,i} - n_i\hat{\theta})^2}{n_i\hat{\theta}(1-\hat{\theta})}$$

Where:

$$\hat{\theta} = \Sigma C_1 / i / \Sigma n_i$$

$$n_i = C_{1,i} + C_{2,i}$$

K = number of samples

# SAMPLE PROBLEM

Suppose that a simple test of mechanical ability is given to 4 groups of school children under the conditions found below. Determine whether there is a difference among the effectiveness of the 4 experimental conditions by testing the null hypothesis, which is that the proportion of successes in the 4 groups should not be significantly different.

| Condition | $C_1$ Failure | $C_2$ Success |
|---|---|---|
| Extensive training before lunch | 8 | 42 |
| Extensive training after lunch | 12 | 18 |
| Minimal training before lunch | 50 | 170 |
| Minimal training after lunch | 9 | 90 |

# SOLUTION

| STEP | INSTRUCTIONS | DISPLAY | INPUT |
|---|---|---|---|
| 1 | Run "DIFF" | Differences Among Proportions | |
| | | E,E to end input, C,C to correct | |
| 2 | Enter case 1 fail, success | Sample 1 enter C1,C2? | 8,42 [RTN] |
| | Enter case 2 | Sample 2 enter C1,C2? | 12,18 [RTN] |
| | Enter case 3 | Sample 3 enter C1,C2? | 50,170 [RTN] |
| | **Error** | Sample 4 enter C1,C2? | 19,19 [RTN] |
| | Call correction routine | Sample 5 enter C1,C2? | C,C [RTN] |
| | This will be displayed: | Sample 4 deleted = 19,19 | |
| | | Sample 4 enter C1,C2? | 9,90 [RTN] |
| 3 | End data input | Sample 5 enter C1,C2? | E,E [RTN] |
| 4 | Real display, use [RTN] to | Chi-square = 16,50082 | [RTN] |
| | see next output. | df = 3 | [RTN] |
| | | Theta = .19799 | [RTN] |
| 5 | End program | Run again, View again, or End? | E [RTN] |

# USER INSTRUCTIONS

| STEP | INSTRUCTIONS | DISPLAY | INPUT |
|---|---|---|---|
| 1 | Run "DIFF" | Differences Among Proportions | |
| | | E,E to end input, C,C to correct | |
| 2 | Enter indicated case: | Sample i enter C1,C2? | C1i,C2i[RTN] |
| | If you made an error: | Sample i enter C1,C2? | C,C [RTN] |
| | This will be displayed: | Sample i deleted = C1i,C2i | |
| | Goto 2 until all data has | | |
| | been entered. | | |
| 3 | End data input | Sample i enter C1,C2? | E,E [RTN] |
| 4 | Read output. User [RTN] to | Chi-square = $\chi^2$ | [RTN] |
| | see next output, [BACK] to | df = df | [RTN]/[BACK] |
| | see previous output. | Theta = $\hat{\theta}$ | [RTN]/[BACK] |
| 5 | Review routine | Run again, View again, or End? | |
| | R = rerun the program step 2 | | R [RTN] |
| | V = review results - step 4 | | V [RTN] |
| | E = end program | | E [RTN] |

# VARIABLE NAMES

| NAME | DESCRIPTION | NAME | DESCRIPTION |
|---|---|---|---|
| C1( ) | Storage of condition 1 | C2( ) | Storage of condition 2 |
| A$ | General use, input C1( ) | B$ | Input C2( ) (Condition 2) |
| X1 | Sum of condition 1 | N1 | Sum of N(i) |
| N | Total N | T | Theta |
| D | Degrees of freedom | C2 | Chi-square value |
| NO | Temporary N(i) | | |

# NOTES AND REFERENCES

Notes: 1. A maximum of 300 samples may be entered. To change this limit, change the dimension statement in line 70.

References: 1. J. Freund, MATHEMATICAL STATISTICS, (Prentice-Hall, 1971).

2. Allen L. Edwards, EXPERIMENTAL DESIGN IN PSYCHOLOGICAL RESEARCH (Rinehart & Co., 1950), p. 74.

This program was derived from the HP-41 Users' Library Solutions Book, "Test Statistics", program DIFFERENCES AMONG PROPORTIONS, p. 9.

# PROGRAM LISTING

```
 10 ! DIFF- Differences
 20 ! among proportions
 30 ! REV 11/01/82
 40 !
 50 DELAY .5
 60 DISP "Differences among proportions
    "
 70 DIM C1(300),C2(300),A$[50],B$[50]
 80 !
 90 ! Correction
100 !
110 DEF FNE                              -Routine to correct error and
                                          decrement counters
120 IF N<1 THEN BEEP 220 @ DISP "Must h
    ave data to delete" @ WAIT 1 @ GOTO
     150
130 DISP "Sample";N;"deleted =";C1(N);"
    ,";C2(N) @ WAIT 1
140 X1=X1-C1(N) @ N1=N1-C1(N)-C2(N) @ N
    =N-1
150 FNE=0 @ END DEF
160 !
170 ! Delay routine
180 !
190 DEF FND                              -Wait for 'RTN' or 'BACK' key.
                                          Return 1 for 'BACK' key
200 A$=KEY$ @ IF A$()CHR$(8) AND A$()CH
    R$(13) THEN 200
210 FND=A$=CHR$(8) @ END DEF
220 DEF FNI(X) = INT(X*10^5+.5)/10^5     -Function to define output
                                          precision
230 !
240 ! Input routine
250 !
260 X1,N1,N=0
270 DISP CHR$(197);" to end input, ";CH
    R$(195);" to correct" @ WAIT 1
280 ON ERROR BEEP 220 @ DISP 'Enter ["E -Error trap- returns warning if
    "],["C"],or [C1,C2]' @ WAIT 1.5 @ G  illegal data is entered
    OTO 290
290 DISP "Sample";N+1;" enter C1,C2 ";
    @ INPUT A$,B$
300 IF UPRC$(A$)="E" THEN 340            -End data entry if 'E' is
                                          entered
310 IF UPRC$(A$)="C" THEN U=FNE @ GOTO   -Call correction routine if 'C'
    290                                   is entered
320 C1(N+1)=VAL(A$) @ C2(N+1)=VAL(B$) @  -Increment counters, store
     N=N+1                                condition values, and cont.
330 X1=X1+C1(N) @ N1=N1+C1(N)+C2(N) @ G
    OTO 290
340 IF N<2 THEN BEEP 220 @ DISP "Please -End routine- verify that N>1
     enter more than 1 sample" @ WAIT 1
     @ GOTO 290
350 !
```

# PROGRAM LISTING

```
360 ! Computation
370 !
380 T=X1/N1                                  -Compute Theta
390 D=N-1
400 C2=0                                     -Loop to compute chi-square
                                              value
410 FOR I=1 TO N
420 N0=C1(I)+C2(I)
430 C2=C2+(C1(I)-N0*T)^2/(N0*T*(1-T))
440 NEXT I
450 !
460 ! Print out
470 !
480 PRINT "Chi-square =";FNI(C2) @ U=FN      -Routine to display data
    D @ IF U THEN 480
490 PRINT "df=";D @ U=FND @ IF U THEN 4
    80
500 PRINT "Theta=";FNI(T) @ U=FND @ IF
    U THEN 490
510 !
520 ! Review routine
530 !
540 DISP CHR$(210);"un again, ";CHR$(21      -Review module
    4);"iew again, or ";CHR$(197);"nd "
    ;
550 INPUT A$ @ A$=UPRC$(A$&" ")
560 ON POS("RVE",A$[1,1])+1 GOSUB 540,2
    60,480,570
570 STOP
```

# PROGRAM DESCRIPTION

## DATA TRANSFORMATIONS

This program will either transform or standardize data sets. The square-root transformation is appropriate in analysis of variance when cell variances tend to be functions of the cell means. The log transformation is useful when normalizing distributions with positive skew.

The standard score (z score) manipulation converts distributions into standard score form, with a mean of zero and unit standard deviation. The T transformation changes data into a distribution with a mean and standard deviation defined by the user.

Formula:

$$\text{mean} = \bar{X} = \frac{\Sigma x}{n}$$

$$\text{Standard deviation } \sigma_x = \sqrt{\left(\frac{\Sigma x^2}{n}\right) - \left(\frac{\Sigma x}{n}\right)^2}$$

$$\text{Log transformation} = \log_e X$$

$$\text{Square root transformation} = \sqrt{X}$$

$$\text{Standard score } s = \frac{X - \bar{X}}{\sigma_x}$$

$$\text{T transform } t = s(\sigma_c) - \bar{X}_c$$

Where: $\sigma_c$ is the constant standard deviation.

$\bar{X}_c$ is the constant mean.

# SAMPLE PROBLEM

The following data represents the time (in minutes) that it took 8 new HP-75 owners to build a short, working BASIC program on their new computers. Convert these scores to:

1) standard scores, and

2) t-scores with a mean of 75 and a standard deviation of 12.

| | Owner | | | | | | | |
|---|---|---|---|---|---|---|---|---|
| | 1 | 2 | 3 | 4 | 5 | 6 | 7 | 8 |
| Time (minutes) | 27 | 15 | 26 | 17 | 12 | 9 | 8 | 27 |

(Artificial data)

# SOLUTION

| STEP | INSTRUCTIONS | DISPLAY | INPUT |
|---|---|---|---|
| | Run "TRANS" | Transformations | |
| | | E to end input, C to correct | |
| 1 | Enter datum 1 | Item 1 Score? | 27 [RTN] |
| | Enter datum 2 | Item 2 Score? | 15 [RTN] |
| | Enter datum 3 | Item 3 Score? | 26 [RTN] |
| | **ERROR** | Item 4 Score? | 177 [RTN] |
| 2 | Call error correction | Item 5 Score? | C [RTN] |
| | This will be displayed | Item 4 deleted = 177 | |
| 1 | Enter correct value | Item 4 Score? | 17 [RTN] |
| | Enter datum 5 | Item 5 Score? | 12 [RTN] |
| | Enter datum 6 | Item 6 Score? | 9 [RTN] |
| | Enter datum 7 | Item 7 Score? | 8 [RTN] |
| | Enter datum 8 | Item 8 Score? | 27 [RTN] |
| 3 | End data input | Item 9 Score? | E [RTN] |
| 4 | Read value of mean and SD. | Mean = 17.625 Std. = 7.51561 | [RTN] |
| 5 | | Enter type of transformation: | |

# SOLUTION

| STEP | INSTRUCTIONS | DISPLAY | INPUT |
|---|---|---|---|
| | Enter transformation type | Log, STandard, SQr, or T? | ST [RTN] |
| 5b | | Std. score transformation | [RTN] |
| 6 | Read transofrmations. Use | Item 1 Raw = 27 Trn = 1.2474 | [RTN] |
| | [RTN] key to read the next | Item 2 Raw = 15 Trn = -.34927 | [RTN] |
| | transformation. | Item 3 Raw = 26 Trn = 1.11435 | [RTN] |
| | | Item 4 Raw = 17 Trn = -.08316 | [RTN] |
| | | Item 5 Raw = 12 Trn = -.74844 | [RTN] |
| | | Item 6 Raw = 9 Trn = -1.14761 | [RTN] |
| | | Item 7 Raw = 8 Trn = -1.28067 | [RTN] |
| | | Item 8 Raw = 27 Trn = 1.2474 | [RTN] |
| 7 | Options: Select 'view' | Run again, View again, or End? | V [RTN] |
| 4 | Read the mean again | Mean = 17.625 Std. = 7.51561 | [RTN] |
| 5 | | Enter type of transformation: | |
| | Enter transformation t | Log, STandard, SQr, or T? | T [RTN] |
| 5a | Enter constant $\mu,\sigma$ | Enter $\mu,\sigma$? | 75,12 [RTN] |
| 5b | | T score transformation | [RTN] |
| 6 | Read the transformations | Item 1 Raw = 27 Trn = 89.96885 | [RTN] |
| | | Item 2 Raw = 15 Trn = 70.80872 | [RTN] |
| | | Item 3 Raw = 26 Trn = 88.37217 | [RTN] |
| | | Item 4 Raw = 17 Trn = 74.00208 | [RTN] |
| | | Item 5 Raw = 12 Trn = 66.01869 | [RTN] |
| | | Item 6 Raw = 9 Trn = 61.22866 | [RTN] |
| | | Item 7 Raw = 8 Trn = 59.63198 | [RTN] |
| | | Item 8 Raw = 27 Trn = 89.96885 | [RTN] |
| 7 | End program | Run again, View again, or End? | E [RTN] |

# USER INSTRUCTIONS

| STEP | INSTRUCTIONS | DISPLAY | INPUT |
|---|---|---|---|
| | Run "TRANS" | Transformations | |
| | | E to end, C to correct | |
| 1 | Enter value of item #i | Item i Score? | Ai [RTN] |
| 2 | If you made an error: | Item i Score? | C [RTN] |
| | This will be displayed | Item i Deleted = $A_i$ | |
| | Goto 1 until all data is in | | |
| 3 | End data input | Item i Score? | E [RTN] |
| 4 | Read mean and std. deviation | Mean = Std. = | [RTN] |
| 5 | Enter transformation type: | Enter type of transformation: | |
| | | Log, STandard, SQr, or T? | |
| | L = Log (base e) transformation | | L [RTN] or |
| | ST = Standard score transformation | | ST [RTN] or |
| | SQ = Square root transformation | | SQ [RTN] or |
| | T = t-transformation | | T [RTN] |
| 5a | If t was chosen: | Enter $\mu,\sigma$? | $\mu,\sigma$ [RTN] |
| 5b | Transformation type is displayed | transformation | [RTN] |
| 6 | Read transformations | Item i Raw = $R_i$ Trn = $T_i$ | [RTN]/[BACK] |
| 7 | Review routine | Run again, View again, or End? | |
| | R = rerun the program - step 1 | | R [RTN] |
| | V = review the answers and | | V [RTN] |
| | select new transformation | | |
| | type - step 4 | | |
| | E = exit the program | | E [RTN] |

# VARIABLE NAMES

| NAME | DESCRIPTION | NAME | DESCRIPTION |
|---|---|---|---|
| A( ) | Data storage | A$ | General use input string |
| T | Type of transformation | N | Total N of data |
| S1 | Sum of all data $\Sigma X_i$ | S2 | Sum of all data-squared $\Sigma(X_i^2)$ |
| K$ | Delay function string | M | Mean of data |
| S | Standard deviation (N) | D | Value of displayed trans. |
| MO | Constant $\mu$ for T | SO | Constant $\sigma$ |
| U | Delay, = 1 if [BACK] used | X | Increment for output |

# NOTES AND REFERENCES

Note: 1. The program is limited to a maximum of 500 items. To change this alter the dimension statement in line 80.

References: 1. B.J. Winer, STATISTICAL PRINCIPLES in Experimental Design (2nd ED), (McGraw-Hill, New York, 1971).

2. G.A. Ferguson, STATISTICAL ANALYSIS in Psychology and Education (2nd ED), (McGraw-Hill, New York, 1966), p. 109.

# PROGRAM LISTING

```
 10 ! TRANS - Data transformation
 20 ! Log base e, square root,
 30 ! standard score, & t score
 40 !
 50 ! REV 11/01/82
 60 !
 70 DISP "          Transformations" @ W
    AIT 1
 80 DIM A(500),A$[60]
 90 INTEGER T
100 DEF FNI(X) = INT(X*10^5+.5)/10^5      -Function to define output
                                           precision
110 DEF FNE                               -Function to handle error
120 IF N<1 THEN BEEP 220 @ DISP "Must h
    ave data to delete" @ WAIT 1 @ GOTO
     150
130 DISP "Item";N;"deleted =";A(N)
140 S1=S1-A(N) @ S2=S2-A(N)^2 @ N=N-1     -Decrement counters
150 FNE=0 @ END DEF
160 !
170 ! Delay function                       -Delay function. Wait for 'RTN'
                                           or 'BACK' keys.
180 !
190 DEF FND
200 K$=KEY$ @ IF K$<>CHR$(13) AND K$<>C
    HR$(8) THEN 200
210 FND=K$=CHR$(8) @ END DEF
220 N,S1,S2=0                             -Beginning of user dialogue
230 DISP CHR$(197);" to end input, ";CH
    R$(195);" to correct" @ WAIT 1
240 ON ERROR BEEP 220 @ DISP "Enter [E]   -Error trap
    [RTN],[C][RTN],or data" @ WAIT 1 @
    GOTO 250
250 DISP "Item";N+1;" Score"; @ INPUT A
    $
260 IF POS(UPRC$(A$),"E") THEN 330        -Check to see if user wishes to
                                           end data entry
270 IF POS(UPRC$(A$),"C") THEN U=FNE @    -Check to see if user wishes to
    GOTO 250                               correct data
280 S1=S1+VAL(A$) @ S2=S2+VAL(A$)^2 @ N   -Increment counters
    =N+1
290 A(N)=VAL(A$) @ GOTO 250
300 !
310 ! End data input
320 !
330 IF N<2 THEN BEEP 220 @ DISP "Must h   -Check validity of sample size
    ave more than 1 item" @ WAIT 1 @ GO
    TO 250
340 M=S1/N @ S=SQR(S2/N-(S1/N)^2)         -Compute mean and standard
                                           deviation
350 PRINT "Mean= ";FNI(M);"SD.= ";FNI(S
    )
360 U=FND @ IF U THEN 350
370 DISP "Enter type of transformation:   -Prompt user to select
    "                                      transformation type
```

# PROGRAM LISTING

```
380 DISP CHR$(204);"og, ";CHR$(211);CHR
    $(212);"andard, ";CHR$(211);CHR$(20
    9);"r, or ";
390 DISP CHR$(212); @ INPUT A$
400 A$=UPRC$(A$&"  ") @ A$=A$[1,2]
410 IF POS("L LOSQSTT ",A$)=0 THEN 370
420 T=POS("L LOSQSTT ",A$) @ IF T>1 THE
    N T=T-1
430 IF T>1 THEN T=T/2
440 IF T=4 THEN DISP "Enter ";CHR$(12);
    ",";CHR$(9); @ INPUT M0,S0
450 A$="Log(e)      Square-rootStd.score
      T-score    "
460 PRINT A$[T*11-10,T*11];" transforma
    tion"
470 U=FND
480 OFF ERROR
490 X=0
500 X=X+1 @ IF X>N THEN 610

510 IF T=1 THEN D=LOG(A(X))
520 IF T=2 THEN D=SQR(A(X))

530 IF T=3 OR T=4 THEN D=(A(X)-M)/S

540 IF T=4 THEN D=D*S0+M0

550 PRINT "Item";X;"Raw=";A(X);"Trn=";F
    NI(D)
560 U=FND @ IF U THEN X=X-1*(X>1) @ GOT
    O 510
570 GOTO 500
580 !
590 ! Review routine
600 !
610 DISP CHR$(210);"un again, ";CHR$(21
    4);"iew again, or ";CHR$(197);"nd";
620 INPUT A$ @ A$=UPRC$(A$&" ")
630 ON POS("RVE",A$[1,1])+1 GOTO 610,22
    0,350,640
640 STOP
```

-Enter mean and SD for t-transformation

-Display name of transformation type

-Increment counter and check for end of data
-Compute log transformation
-Compute square-root transformation
-Compute t and standard score transformation
-Transform mean and SD for t-transformation
-Display transformed data

-Program options

## TEST STATISTICS

ONE SAMPLE TEST STATISTICS FOR THE MEAN
KENDALL'S COEFFICIENT OF CONCORDANCE
CORRELATION COEFFICIENT TEST
INTRACLASS CORRELATION COEFFICIENT
KRUSKAL-WALLIS STATISTIC
MANN-WHITNEY U-TEST
FISHER'S EXACT PROBABILITY
2-FACTOR ANALYSIS OF VARIANCE
BARTLETT'S CHI-SQUARE STATISTIC
DIFFERENCES AMONG PROPORTIONS
DATA TRANSFORMATIONS

ALL HP-75 SOLUTIONS BOOKS ARE AVAILABLE RECORDED ON MINI-DATA CASSETTES FROM EITHER A HEWLETT-PACKARD DEALER OR THE HP USERS' LIBRARY.


00075-90076 Printed in U.S.A.

Reorder No.
00075-13012